code:
;-------------------------------------------------
;アイテム画面
;-------------------------------------------------
@COM490
#DIM 대상
#DIM 전원
#DIM CLIENT_ID
#DIM 공허로회귀
#DIM 사용아이템
#DIM 표시아이템종별
#DIMS 식품특징
VARSET LOCAL
VARSET LOCALS
대상 = MASTER
공허로회귀 = 0
PRINTL 무엇을 사용합니까？
$LOOP
DRAWLINE
FOR LOCAL,120, 699
	SIF 표시아이템종별 != 0 && ITEMTYPE(LOCAL) != 표시아이템종별
		CONTINUE
	IF 대상 != MASTER
		IF 사용가능아이템(LOCAL) == 2 && ITEM:LOCAL
			CALL PRINT_ITEM_EXPLANATION(LOCAL)
		ENDIF
	ELSE
		IF 사용가능아이템(LOCAL) && ITEM:LOCAL
			CALL PRINT_ITEM_EXPLANATION(LOCAL)
		ENDIF
	ENDIF
NEXT
PRINTL
DRAWLINE
CALL PRINTBUTTON_EX("[전체표시]", 2000, 표시아이템종별 != 0, 표시아이템종별 == 0)
CALL PRINTBUTTON_EX("[ 회복약 ]", 2001, 표시아이템종별 != 1, 표시아이템종별 == 1)
CALL PRINTBUTTON_EX("[ 버프약 ]", 2002, 표시아이템종별 != 2, 표시아이템종별 == 2)
CALL PRINTBUTTON_EX("[ 식　품 ]", 2003, 표시아이템종별 != 3, 표시아이템종별 == 3)
CALL PRINTBUTTON_EX("[　과자　]", 2004, 표시아이템종별 != 4, 표시아이템종별 == 4)
CALL PRINTBUTTON_EX("[ 과　일 ]", 2005, 표시아이템종별 != 5, 표시아이템종별 == 5)
CALL PRINTBUTTON_EX("[ 그　외 ]", 2006, 표시아이템종별 != 6, 표시아이템종별 == 6)
RESETCOLOR
PRINTL
DRAWLINE
SIF ITEM:곤충채집세트
	PRINTFORML [901] 곤충 채집통을 확인한다
PRINTFORML [900] 콜렉션의 스키마 자동 사양 변환（현재\@ FLAG:C스키마자동사양 ? ON# OFF\@）
IF TARGET && SHIRAHU(TARGET)
	SIF 대상 != MASTER
		SETCOLOR C_YELLOW
	PRINTFORML [998] %CALLNAME:TARGET%에 사용한다
	RESETCOLOR
ENDIF

PRINTL [999] 그만둔다
INPUT
사용아이템 = RESULT
;アイテム以外の処理
SELECTCASE RESULT
	CASE 999
		RETURN -1
	CASE 900
		FLAG:C스키마자동사양 = !FLAG:C스키마자동사양
		PRINTFORMW 자동 사양을\@ FLAG:C스키마자동사양 ? ON# OFF\@(으)로 했습니다
		GOTO LOOP
	CASE 901
		;[虫捕りをする]へ転送
		CALL SHOW_MUSHI_CAGE
		GOTO LOOP
	CASE 998
		IF 대상 != MASTER
			대상 = MASTER
		ELSEIF SHIRAHU(TARGET) && TARGET
			대상 = TARGET
		ENDIF
		GOTO LOOP
	CASE 2000 TO 2006
		표시아이템종별 = RESULT % 2000
		GOTO LOOP
ENDSELECT

;使用可能アイテム以外の入力、またはアイテム未所持の場合は再入力
IF 사용가능아이템(사용아이템) == 0 || ITEM:사용아이템 <= 0
	GOTO LOOP
ENDIF
;相手使用時にMASTERのみアイテムの選択は再入力
IF 대상 != MASTER && 사용가능아이템(사용아이템) == 1
	GOTO LOOP
ENDIF
;IF ITEMTYPE_DRUG(使用アイテム,対象) && TCVAR:(ARG:1):お薬
; 	PRINTL 이미 약을 복용했으니 오늘은 이제 그만두자
; 	GOTO LOOP
;ENDIF
IF ITEMTYPE_FOOD(사용아이템) && !TIME_PROGRESS(TCVAR:대상:공복시각)
	PRINTFORML %조사처리(CALLNAME:대상,"은")% 아직 배가 고프지 않다
	GOTO LOOP
ENDIF
IF ITEMTYPE_SWEETS(사용아이템) && !TIME_PROGRESS(TCVAR:대상:디저트공복시각)
	PRINTL 간식은 아직 먹고 싶지 않다
	GOTO LOOP
ENDIF
TFLAG:193 = 사용아이템
TFLAG:194 = 대상 == MASTER ? MASTER # 1
SELECTCASE 사용아이템
	CASE 120	;블랙썬더
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		SIF 대상 == MASTER
			PRINTL ……싸구려 맛이 났다
		식품특징 = %ITEMNAME:사용아이템%/단/정키/사각사각/
		CALL RECOVER(대상, ITEM_TASTE(대상, 50, 식품특징), "체력", "초코에 의한 회복")
		CALL RECOVER(대상, ITEM_TASTE(대상, 50, 식품특징), "기력", "초코에 의한 회복")
		CALL 만복도상승(대상, "디저트", 60)
		LOCAL:1 ++
	CASE 121	;초콜릿
		IF 대상 == MASTER
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
			식품특징 = %ITEMNAME:사용아이템%/단/
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 300, 식품특징), "체력", "초코에 의한 회복")
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 300, 식품특징), "기력", "초코에 의한 회복")
			CALL 만복도상승(대상, "디저트")
			LOCAL:1 ++
		ENDIF
	CASE 122	;고급초콜릿
		IF 대상 == MASTER
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
			PRINTFORML 맛있어!
			식품특징 = %ITEMNAME:사용아이템%/단/품위있는/호화/고급식재/
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 500, 식품특징), "체력", "초코에 의한 회복")
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 500, 식품특징), "기력", "초코에 의한 회복")
			CALL 만복도상승(대상, "디저트")
			LOCAL:1 ++
		ENDIF
	CASE 123	;진심초콜릿
		IF 대상 == MASTER
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
			PRINTFORML 사랑이 느껴져!
			식품특징 = %ITEMNAME:사용아이템%/애정/단/
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 1000, 식품특징), "체력", "초코에 의한 회복")
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 1000, 식품특징), "기력", "초코에 의한 회복")
			CALL 만복도상승(대상, "디저트")
			LOCAL:1 ++
		ENDIF
	CASE 212	;떡잎으로싼떡
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		LOCAL:1 ++
		;端午の節句は여름의季語だそうなので봄철じゃなかった
		식품특징 = %ITEMNAME:사용아이템%/단/
		CALL RECOVER(대상, ITEM_TASTE(대상, 100, 식품특징), "체력", "떡잎으로 싼 떡에 의한 회복")
		CALL RECOVER(대상, ITEM_TASTE(대상,   0, 식품특징), "기력", "떡잎으로 싼 떡에 의한 회복")
		CALL 만복도상승(대상, "디저트")
	CASE 213	;아이스
		SIF TARGET && SHIRAHU(TARGET)
			PRINT 둘이 나란히
			PRINTDATA
				DATAFORM 「아이스」
				DATAFORM 「냉동 귤」
				DATAFORM 「(무늬)격이 2개 붙은 소다 아이스」
				DATAFORM 「2개(살)로 1조의 달걀 아이스」
				DATAFORM 「츄-애완동물」
				DATAFORM 「파피코」
				DATAFORM 「연유가 찬 milk bar」
				DATAFORM 「아즈 어금니-」
				DATAFORM 「수박 바」
				DATAFORM 「가튼, 이라고 귤」
				DATAFORM 「블랙 번개 아이스」
				DATAFORM 「눈구경 큰 복」
			ENDDATA
		PRINTFORML (을)를 먹었다
		식품특징 = %ITEMNAME:사용아이템%/차가운/단/
		CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 150, 식품특징), "기력", "아이스크림에 의한 회복")
		CALL 만복도상승(MASTER, "디저트")
			IF  TARGET && SHIRAHU(TARGET)
			CALL RECOVER(TARGET, ITEM_TASTE(TARGET, 150, 식품특징), "기력", "아이스크림에 의한 회복")
			CALL 만복도상승(TARGET, "디저트")
			ENDIF
		LOCAL:1 ++
	CASE 215	;홍차
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 사용했다
		SETBIT TCVAR:MASTER:특별한차, 0
		LOCAL:1 ++
	CASE 352	;딸기
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		식품특징 = %ITEMNAME:사용아이템%/야채/과일/단/
		CALL RECOVER(대상, ITEM_TASTE(대상, 50, 식품특징), "기력", "딸기에 의한 회복")
		LOCAL:1 ++
	CASE 364	;수박
		식품특징 = %ITEMNAME:사용아이템%/야채/과일/단/
		IF FLAG:70 == 1
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
			CALL RECOVER(대상, ITEM_TASTE(대상, 100, 식품특징), "기력", "수박에 의한 회복")
		ELSE
			IF TARGET != MASTER
				IF GET_TARGETNUM() > 1
					PRINTFORM 다같이
				ELSEIF SHIRAHU(TARGET)
					PRINTFORM %타겟과%
				ENDIF
			ENDIF
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
			CALL RECOVER(MASTER, ITEM_TASTE(MASTER, 100, 식품특징), "기력", "수박에 의한 회복")
			FOR 전원, 1, GET_TARGETNUM() + 1
				SIF TARGET:전원 <= 0
					CONTINUE
				SIF !SHIRAHU(전원)
					CONTINUE
				CALL RECOVER(전원, ITEM_TASTE(대상, 100, 식품특징), "기력", "수박에 의한 회복")
			NEXT
		ENDIF
		LOCAL:1 ++
	;CASE 400
	;	PRINTFORML %CALLNAME:MASTER%は%ITEMNAME:사용아이템%を開いた
	;	CALL COM401
	;	LOCAL:1 ++
	CASE 500	;녹즙
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		PRINTL 맛없어! 한잔 더!
		; 한판 주, 맛 데이터에 苦い가 없는데 苦い를 쓰고 있음. 일단 주의
		식품특징 = %ITEMNAME:사용아이템%/苦い/苦い/苦い/건강한/야채/이파리/야취/
		CALL RECOVER(대상,100, "체력", "녹즙에 의한 회복")
		CALL RECOVER(대상, ITEM_TASTE(대상, 200, 식품특징), "기력", "녹즙에 의한 회복")
		LOCAL:1 ++
	CASE 501	;영양드링크
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		CALL RECOVER(대상, 약학지식(200), "체력", "영양드링크에 의한 회복")
		CALL RECOVER(대상, 약학지식(100), "기력", "영양드링크에 의한 회복")
		CFLAG:대상:졸음쫒기강도 = MAX(약학지식(20), CFLAG:대상:졸음쫒기강도)
		TCVAR:대상:약 = 1
			LOCAL:1 ++
	CASE 502	;정력제
		IF HAS_PENIS(대상)
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
			CALL RECOVER(대상, 약학지식(300), "정력", "정력제에 의한 회복")
			TCVAR:대상:약 = 1
			LOCAL:1 ++
		ELSE
			PRINTFORML 정력제의 사용 대상이 아닌 것 같다
			GOTO LOOP
		ENDIF
	CASE 503	;불끈불끈Z
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		PRINTL 파이팅 한발, 가득！
		CALL RECOVER(대상, 약학지식(500), "체력", "불끈불끈Z에 의한 회복")
		CALL RECOVER(대상, 약학지식(300), "기력", "불끈불끈Z에 의한 회복")
		CFLAG:대상:졸음쫒기강도 = MAX(약학지식(60), CFLAG:대상:졸음쫒기강도)
		TCVAR:대상:약 = 1
		LOCAL:1 ++
	;CASE 504
	;		IF FLAG:70
	;			PRINTFORML 時間停止を解除してから使おう
	;			GOTO LOOP
	;		ENDIF
	;		IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
	;			PRINTFORML これは自室で飲もう
	;			GOTO LOOP
	;		ENDIF
	;		PRINTFORML ストロング虚無を口にした途端、虚無の力で思考はぼやけ、すべてがどうでもよくなってきた
	;		PRINTFORML %CALLNAME:MASTER%は缶を空けながら一日を無為に過ごした…
	;		공허로회귀 = 1
	CASE 505	;반짝반짝Ｘ
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		PRINTL 24시간 싸울 수 있을 것 같은 생각이 든다…！
		CALL RECOVER(대상, 약학지식(800), "체력", "반짝반짝Ｘ에 의한 회복")
		CALL RECOVER(대상, 약학지식(600), "기력", "반짝반짝Ｘ에 의한 회복")
		CFLAG:대상:졸음쫒기강도 = MAX(약학지식(180), CFLAG:대상:졸음쫒기강도)
		CFLAG:대상:철야 = 1
		TCVAR:대상:약 = 1
			LOCAL:1 ++
	CASE 506	;정력제『천원凸파』
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		PRINTL 너의××× 그리고 하늘을 찔러라!
		CALL RECOVER(대상, 약학지식(400), "체력", "정력제『천원凸파』에 의한 회복")
		CALL RECOVER(대상, 약학지식(300), "기력", "정력제『천원凸파』에 의한 회복")
		CALL RECOVER(대상, 약학지식(800), "정력", "정력제『천원凸파』에 의한 회복")
		TCVAR:대상:약 = 1
		LOCAL:1 ++
	CASE 507	;포션
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		CALL RECOVER(대상, 약학지식(150), "기력", "포션에 의한 회복")
		CALL RECOVER(대상, 약학지식(150), "TSP", "포션에 의한 회복")
		TCVAR:대상:약 = 1
		LOCAL:1 ++
	CASE 508	;매직포션
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		PRINTL 마력이 넘쳐 온다…
		CALL RECOVER(대상, 약학지식(300), "기력", "매직포션에 의한 회복")
		CALL RECOVER(대상, 약학지식(600), "TSP", "매직포션에 의한 회복")
		TCVAR:대상:약 = 1
		LOCAL:1 ++
	CASE 509	;잠깨기포션
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		CALL RECOVER(대상, 약학지식(200), "기력", "잠깨기포션에 의한 회복")
		CFLAG:대상:졸음쫒기강도 = MAX(약학지식(240), CFLAG:대상:졸음쫒기강도)
		CFLAG:대상:철야 = 1
		TCVAR:대상:약 = 1
		LOCAL:1 ++
	CASE 510	;우콘드링크
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
			PRINTL 간기능이 높아져, 취하기 어려워졌다
			TFLAG:멀미약 = 1
			LOCAL:1 ++
	CASE 511	;엘릭서
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		PRINTL 피로가 저 쪽으로 물리쳤다…！
		CALL RECOVER(대상, MAX(9999, MAXBASE:대상:체력), "체력", "엘릭서에 의한 회복")
		CALL RECOVER(대상, MAX(9999, MAXBASE:대상:기력), "기력", "엘릭서에 의한 회복")
		CALL RECOVER(대상, MAX(9999, MAXBASE:대상:정력), "정력", "엘릭서에 의한 회복")
		CALL RECOVER(대상, MAX(9999, MAXBASE:대상:TSP),  "TSP",  "엘릭서에 의한 회복")
		CFLAG:대상:졸음쫒기강도 = MAX(300, CFLAG:대상:졸음쫒기강도)
		TCVAR:대상:약 = 1
		LOCAL:1 ++
	CASE 512	;선단
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
			PRINTL 심상치 않은 힘이 넘친다……！
			CALL BUFF_BASE(대상, BASE_체력, MAX(MAXBASE:대상:체력 / 5, 500))
			CALL BUFF_BASE(대상, BASE_기력, MAX(MAXBASE:대상:기력 / 5, 500))
			CALL BUFF_BASE(대상, BASE_정력, MAX(MAXBASE:대상:정력 / 5, 500))
			LOCAL:1 ++
	CASE 514	;선향옥토
		IF TFLAG:선향옥토 > 0
			PRINTFORML 이미 사용하고 있다
			GOTO LOOP
		ELSE
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 향로에 지폈다
			PRINTL 현혹적인 훈향이 주위에 감돈다……
			TFLAG:선향옥토 = 240
			LOCAL:1 ++
		ENDIF
	CASE 515	;국사무쌍약　３回目まではバフ効果
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 마셨다
		IF TCVAR:대상:국사무쌍약 < 3
			PRINTL 힘이 넘쳐 흐른다……！
			CALL BUFF_BASE(대상, BASE_체력, MAX(MAXBASE:대상:체력 / 15, 150))
			CALL BUFF_BASE(대상, BASE_정력, MAX(MAXBASE:대상:정력 / 10, 150))
			TCVAR:대상:국사무쌍약 ++
			LOCAL:1 ++
		ELSE
			;四回目で爆発、デバフに
			SETCOLOR C_RED
			PRINTFORMW ！！
			PRINTFORML ……폭발해서 아프로 머리가 되었다……너무 많이 마신 것 같다……
			RESETCOLOR
			CALL BUFF_BASE(대상, BASE_체력, MIN((-1 * (MAXBASE:대상:체력 / 5)), -450))
			CALL BUFF_BASE(대상, BASE_정력, MIN((-1 * (MAXBASE:대상: 정력 / 3)), -450))
			TCVAR:대상:국사무쌍약 = 0
			LOCAL:1 ++
		ENDIF
	CASE 516	;가싱가든
		IF TFLAG:GG설치위치 == CFLAG:MASTER:현재위치
			PRINTFORMW %TEXTR(@"「제가 키웠습니다」（%GET_MAPNAME(MAIN_MAP)%거주　%CALLNAME:MASTER%씨）/본 품종은 안정성을 고려하고 있지만, 만일 불편하신 분이 있다면 거리를 두고 충분한 휴식을 취해 주세요")%
		ELSEIF TFLAG:GG설치위치
			PRINTFORML 가싱가든은 하루 한도 덧없는 생명
			PRINTFORMW 한 그루를 최후까지 지켜보자
			GOTO LOOP
		ELSEIF INROOM(CFLAG:MASTER:현재위치)
			PRINTFORMW 야외에서 하자
			GOTO LOOP
		;目撃者がいると駄目
		ELSEIF (TARGET > 0 || DATE_HITOGOMI(CFLAG:MASTER:현재위치) > 0) && !FLAG:70
			PRINTFORMW 여기서는 남의 눈이 있다……몰래 장치해야 할 것이다
			GOTO LOOP
		ELSE
			PRINTFORML %마스터는% 마법의 씨앗을 뿌렸다
			PRINTFORML 순식간에 마법 생물이 피어나기 시작해, 주위에 정욕을 돋구는 향기가 가득하다……
			TFLAG:GG설치위치 = CFLAG:MASTER:현재위치
			LOCAL:1 ++
		ENDIF
	CASE 517	;포이즌바디
		IF TFLAG:포이즌바디 > 0
			PRINTFORML 이 이상 섭취하면 간 기능에 장애가 온다
			GOTO LOOP
		ELSEIF HAS_PENIS(대상)
			PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 복용했다
			PRINTFORML 빙글빙글 몸 속에서 소용돌이치는 뭔가의 감각과 함께, 거기가 뜨거워졌다……
			PRINTFORML （%CALLNAME:MASTER%의 정액에 발정 작용 부여　효과발휘 중에는 사정할 때 체력・기력소비 증가）
			PRINTFORML %마스터는% 두통・현기증・구역질・관절통이 업습했지만, 건강보다 중요한 것도 있다며 마음을 굳게 먹고 견뎠다
			TFLAG:포이즌바디 = 240
			LOCAL:1 ++
		ELSE
			PRINTFORML 이 약은 남성 전용인 것 같다
			GOTO LOOP
		ENDIF
	CASE 520	;붕대
		IF !TFLAG:키스마크
			PRINTFORML 지금 사용해도 의미가 없다
			GOTO LOOP
		ELSE
			PRINTFORML %마스터는% %조사처리(ITEMNAME:사용아이템,"를")% 감아 키스 마크를 숨겼다
			TFLAG:붕대 = 1
			LOCAL:1 ++
		ENDIF
	CASE 521	;역서
		IF FLAG:70
			PRINTFORML 시간 정지를 해제하고 나서 사용하자
			GOTO LOOP
		ENDIF
		PRINTFORML %마스터는% 역서를 열었다
		PRINTFORML 점쟁이가 나왔다
		PRINTFORMW
		IF CFLAG:[[레이무]]:현재위치 == CFLAG:MASTER:현재위치
			PRINTFORML 재빠르게 레이무는 점쟁이의 머리에 불제봉을 내려쳤다.
			PRINTFORML 가련한 점쟁이는 머V리 가 되서 떠나갔다.
			FOR LOCAL,1,CHARANUM
				IF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
					BASE:LOCAL:이성 = MAXBASE:LOCAL:이성
					BASE:LOCAL:무드 = 0
				ENDIF
			NEXT
			LOCAL:1 ++
		ELSE
			PRINTFORML 「점술을 통해서 가야할 길을 봐 주겠소」
			$EKISYA
			CALL ASK_M("오늘의 운세", !GETBIT(FLAG:999, 1), "곤란해 하고있는 사람을 찾는다.", 1, "손톱깍이를 대령하라", BASE:MASTER:TSP >= 20, "미라다의돌 해주(100카리스마)", MONEY:2 >= 100, "엄지척!", 1, "공물로 바친다", 1)
			SELECTCASE RESULT
				CASE 0
					SELECTCASE RAND:10
						CASE 0
							TFLAG:행운보정 = 3
							PRINTFORML 「절호조다, 복권이라도 뽑으러 가는게 어때？」
						CASE 1 TO 3
							TFLAG:행운보정 = 1
							PRINTFORML 「양호하다, 미루고 있던 일이 있으면 지금 해라」
						CASE 4,5
							PRINTFORML 「좋지도, 나쁘지도 않음, 이라는 느낌이다.」
						CASE 6 TO 8
							TFLAG:행운보정 = -1
							PRINTFORML 「저조하다, 무리는 하지 말아라」
						CASE 9
							TFLAG:행운보정 = -3
							PRINTFORML 「...너무 안좋다, 오늘 하루는 얌전히 있어라」
					ENDSELECT
				CASE 1
					CLIENT_ID = SEARCH_UNIQUE_COM_사람돕기()
					IF !CLIENT_ID
						PRINTFORML 「그런 사람은 딱히 보이지 않는구나」
						PRINTFORML 「굳이 말하자면 나다, 홍백 무녀에게 노려지고 있다」
						PRINTFORML ...그건 어떻게 해줄수도 없을 것 같다.
					ELSE
						PRINTFORML 「%CALLNAME:CLIENT_ID% 인가?, 저 아이가 도움을 요구하고 있는 것 같구나」
					ENDIF
				CASE 2
					PRINTFORMW %마스터가% 점쟁이를 가만히 바라보자 눈길이 되돌아 왔다.
					PRINTFORMW ……뭐야 그 손톱은.
					PRINTFORMW 인간을 그만두었다는 어필일진 몰라도, 너무 길다.
					PRINTFORMW 일상 생활에서 불편하지 않나?, 타인을 실수로 긁어버리면 어떻게 하려는 거지?
					PRINTFORMW 자신은 언제 부드러운 부분에 손대도 괜찮게 매일 체크하고 있는데...
					PRINTFORMW 아니아니 애초에 무언가에 걸려 생살을 뜯어지게하면 아프지 않은가...
					PRINTFORMW 손톱 밑에 때가 껴도 불편하지 않은건가?
					PRINTFORMW 참을 수 없게 된 %마스터는% 시간을 멈추어 점쟁이의 손톱을 정중하게 잘라 가지런히 했다
					PRINTFORMW %마스터는% 큰 달성감을 맛보았다……
					PRINTFORML 한편 돌연 손톱이 짧아진 점쟁이는 곤혹하고 있다……
					CALL RECOVER_PERMIL(대상,40,"기력")
					BASE:MASTER:TSP -= 20
				CASE 3
					PRINTFORML 누구의 것을 해주 할까?
					FOR LOCAL,1,CHARANUM
						SIF CFLAG:LOCAL:몸목적
							PRINTFORMLC [{LOCAL}] %CALLNAME:LOCAL%
					NEXT
					PRINTFORML
					DRAWLINE
					PRINTFORML [999] 그만둔다
					INPUT
					IF RESULT == 999
						GOTO EKISYA
					ELSEIF CFLAG:RESULT:몸목적
						PRINTFORML 「이런 도구는 좋지 않다고 생각한다」
						PRINTFORMW 점쟁이가 무엇인가 복잡한 의식을 거행하자, 미라다의 돌은 서서히 빛을 잃어 바스라졌다.
						TIME += 25
						CFLAG:RESULT:몸목적 = 0
						MONEY:2 -= 100
					ENDIF
				CASE 4
					PRINTFORML %마스터는% 점쟁이와 시선을 주고 받고는 무언으로 엄지척을 했다
					PRINTFORML 점쟁이는 잠깐 주저하다가, 강력한 엄지척 돌려주어 왔다
					PRINTFORMW 두 사람 사이에 우정이 자라났다.
					GOTO EKISYA
				CASE 5
					PRINTFORML %마스터는% 시간을 멈춰서 점쟁이의 옷을 벗긴 뒤, 속옷을 신께 공물로 바쳤다.
					CALL BUFF_BASE(MASTER,BASE_TSP,100,1)
					PRINTFORMW 자신도 모르는 사이에 속옷을 탈취당한 점쟁이는 당황하고 있다.
			ENDSELECT
			PRINTFORMW
			DRAWLINE
			$EKISYA_DIVIDE
			PRINTFORML 레이무가 나타나 점쟁이의 머리를 불제봉으로 내리쳤다.
			PRINTFORML 가련한 점쟁이는 머리가 2개로 갈라져 머V리가 됐다.
			PRINTFORMW 레이무는 %마스터를% 잠시 바라보더니 떠나갔다.
			FOR LOCAL,1,CHARANUM
				IF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
					BASE:LOCAL:이성 = MAXBASE:LOCAL:이성
					BASE:LOCAL:무드 = 0
				ENDIF
			NEXT
			LOCAL:1 ++
		ENDIF
	CASE 530	;미라다의돌
		CALL SET_MIRADA
		IF RESULT == -1
			RETURN -1
		ELSE
			LOCAL:1 ++
		ENDIF
	CASE 550	;야트메꼬치
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		CALL 만복도상승(대상,"경식")
		식품특징 = %ITEMNAME:사용아이템%/생선 요리/안주/
		CALL RECOVER_PERMIL(대상, ITEM_TASTE(대상, 40, 식품특징), "체력", 0, "야트메꼬치에 의한 회복")
		CALL RECOVER_PERMIL(대상, ITEM_TASTE(대상, 80, 식품특징), "기력", 0, "야트메꼬치에 의한 회복")
		CALL RECOVER_PERMIL(대상, ITEM_TASTE(대상, 50, 식품특징), "정력", 0, "야트메꼬치에 의한 회복")
		LOCAL:1 ++
	CASE 576	;(酒造パッチ)아쿠마키
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		CALL 만복도상승(대상,"주식")
		식품특징 = %ITEMNAME:사용아이템%/일식/쌀/경단/단/쫄깃한/가정적/봄철/
		CALL RECOVER(대상, ITEM_TASTE(대상, 150, 식품특징), "체력", "아쿠마키에 의한 회복")
		CALL RECOVER(대상, ITEM_TASTE(대상, 150, 식품특징), "기력", "아쿠마키에 의한 회복")
		LOCAL:1 ++
	CASE 626	;나무딸기
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		식품특징 = %ITEMNAME:사용아이템%/과일/단/시큼한/야취/
		CALL RECOVER(대상, ITEM_TASTE(대상, 50, 식품특징), "기력", "나무딸기에 의한 회복")
		LOCAL:1 ++
	CASE 628	;으름덩굴
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		식품특징 = %ITEMNAME:사용아이템%/과일/단/야취/
		CALL RECOVER(대상,80, "체력", "으름덩굴에 의한 회복")
		CALL RECOVER(대상, ITEM_TASTE(대상, 80, 식품특징), "기력", "으름덩굴에 의한 회복")
		LOCAL:1 ++
	CASE 630	;선도
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		PRINTL 힘이 넘친다……！
		CALL BUFF_BASE(대상, BASE_체력, MAX(MAXBASE:대상:체력 / 10, 500))
		CALL BUFF_BASE(대상, BASE_기력, MAX(MAXBASE:대상:기력 / 10, 500))
		CALL BUFF_BASE(대상, BASE_정력, MAX(MAXBASE:대상:정력 / 10, 500))
		LOCAL:1 ++
	CASE 665	;(酒造パッチ)머루
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		식품특징 = %ITEMNAME:사용아이템%/과일/단/시큼한/야취/
		CALL RECOVER(대상, ITEM_TASTE(대상, 50, 식품특징), "기력", "머루에 의한 회복")
		LOCAL:1 ++
	CASE 669	;(酒造パッチ)휴대용타코
		PRINTFORML %조사처리(ITEMNAME:사용아이템,"를")% 먹었다
		CALL 만복도상승(대상,"경식")
		;KR 주, 현재 タマネギ(양파)의 맛 데이터는 없음
		식품특징 = 양파/토마토/양배추/주식/야채/짐승고기/스파이스/치즈/매운/푸짐/농후/기능적/야취/선명한/사각사각/
		CALL RECOVER(대상, ITEM_TASTE(대상, 120, 식품특징), "체력", "타코에 의한 회복")
		CALL RECOVER(대상, ITEM_TASTE(대상, 120, 식품특징), "기력", "타코에 의한 회복")
		LOCAL:1 ++
	CASEELSE
		GOTO LOOP
ENDSELECT
IF LOCAL:1
	IF !FLAG:70
		TIME += 5
	ELSE
		BASE:MASTER:TSP -= 5
	ENDIF
	ITEM:사용아이템 -= 1
	SIF 공허로회귀
		CALL OYASUMI
	RETURN 0
ENDIF

@SET_MIRADA
PRINTFORML 대상을 선택해 주세요.
PRINTFORML 황색은 이미 유효, 핑크는 이미 함락 소질이 붙어 있는 캐릭터입니다.
SIF FLAG:정렬
	CALL CHARASORT_EX
FOR LOCAL:1, 1, EX_CHARANUM + 1
	IF FLAG:정렬
		LOCAL = RESULT:(LOCAL:1)
		SIF LOCAL < 0
			CONTINUE
	ELSE
		LOCAL = LOCAL:1
		SIF LOCAL >= CHARANUM
			BREAK
	ENDIF
	;SIF TALENT:LOCAL:스침
	;	CONTINUE
	SIF CFLAG:LOCAL:출금
		CONTINUE
	SIF CFLAG:LOCAL:몸목적
		SETCOLOR C_YELLOW
	SIF TALENT:LOCAL:섹프 || TALENT:LOCAL:연모
		SETCOLOR C_PINK
	PRINTFORMLC [{LOCAL}] %CALLNAME:LOCAL%
	RESETCOLOR
NEXT
PRINTFORML
DRAWLINE
PRINTFORML [999] 그만둔다
INPUT
SIF !INRANGE(RESULT,1,CHARANUM)
	RETURN -1
SIF CFLAG:RESULT:몸목적
	RETURN -1
IF TALENT:RESULT:섹프 || TALENT:RESULT:연모
	TCVAR:RESULT:발정 = 1
ELSE
	CFLAG:RESULT:몸목적 = 1
ENDIF
PRINTFORMW %조사처리(CALLNAME:RESULT,"를")% 떠올리면서 미라다의 돌을 강하게 잡자, 돌은 희미하게 빛을 띠었다
RETURN 1

@사용가능아이템(ARG)
;1だとMASTERのみ、2だとあなた以外にも使える
#FUNCTION

SIF CFLAG:MASTER:우후후 && !ITEMTYPE_DRUG(ARG)
	RETURNF 0
;(酒造パッチ)あくまきと머루、タコス
;SELECTCASE ARG
;	CASE 121 TO 123, 215, 364, 500, 510, 514, 515, 516, 517, 520, 521, 530
;		RETURNF 1
;	CASE 120, 212, 213, 352, 501, 503, 505 TO 509, 511, 512, 550, 576, 626, 628, 630, 665, 669
;		RETURNF 2
;	CASE 502
;		SIF HAS_PENIS(TARGET)
;			RETURNF 2
;		RETURNF 1
;	CASEELSE
;		RETURNF 0
;ENDSELECT
SELECTCASE ITEMTYPE(ARG)
	CASE 1, 2, 3, 4, 5
		IF GROUPMATCH(ITEMNAME:ARG, "국사무쌍약", "초콜릿", "고급초콜릿", "진심초콜릿", "녹즙")
			RETURNF 1
		ENDIF
		RETURNF 2
	CASE 6
		RETURNF 1
	CASEELSE
		RETURNF 0
ENDSELECT

;本来@ITEMDATAを使うべきだが#FUNCTIONじゃないのがなんかやだ
@ITEMTYPE(ARG)
#FUNCTION
IF ITEMTYPE_DRUG(ARG)
	RETURNF 1
ELSEIF ITEMTYPE_BUFF(ARG)
	RETURNF 2
ELSEIF ITEMTYPE_FOOD(ARG)
	RETURNF 3
ELSEIF ITEMTYPE_SWEETS(ARG)
	RETURNF 4
ELSEIF ITEMTYPE_FRUIT(ARG)
	RETURNF 5
ELSEIF ITEMTYPE_ETCETERA(ARG)
	RETURNF 6
ENDIF
RETURNF 0

@ITEMTYPE_DRUG(ARG)
#FUNCTION
;SIF GROUPMATCH(ARG, 501, 502, 503, 505, 506, 507, 508, 509, 511)
SIF GROUPMATCH(ITEMNAME:ARG, "영양드링크", "정력제", "불끈불끈Z", "반짝반짝Ｘ", "정력제『천원凸파』", "포션", "매직포션", "잠깨기포션", "엘릭서")
	RETURNF 1

;BUFF_BASEを行うもの
@ITEMTYPE_BUFF(ARG)
#FUNCTION
;SIF GROUPMATCH(ARG, 512, 515, 630)
SIF GROUPMATCH(ITEMNAME:ARG, "선단", "국사무쌍약", "선도")
	RETURNF 1

;満腹度が増えるもの
@ITEMTYPE_FOOD(ARG)
#FUNCTION
;SIF GROUPMATCH(ARG, 550, 576, 669)
SIF GROUPMATCH(ITEMNAME:ARG, "야트메꼬치", "아쿠마키", "휴대용타코")
	RETURNF 1

;デザート満腹度が増えるもの
@ITEMTYPE_SWEETS(ARG)
#FUNCTION
;SIF GROUPMATCH(ARG, 120, 121, 122, 123, 212, 213)
SIF GROUPMATCH(ITEMNAME:ARG, "블랙썬더", "초콜릿", "고급초콜릿", "진심초콜릿", "떡잎으로싼떡", "아이스크림")
	RETURNF 1

;満腹度が増えないもの
@ITEMTYPE_FRUIT(ARG)
#FUNCTION
;SIF GROUPMATCH(ARG, 352, 364, 500, 626, 628, 665)
SIF GROUPMATCH(ITEMNAME:ARG, "딸기", "수박", "녹즙", "나무딸기", "으름덩굴", "머루")
	RETURNF 1

;その他のもの
@ITEMTYPE_ETCETERA(ARG)
#FUNCTION
;SIF GROUPMATCH(ARG, 215, 510, 514, 516, 517, 520, 521, 530)
SIF GROUPMATCH(ITEMNAME:ARG, "홍차", "우콘드링크", "선향옥토", "가싱가든", "포이즌바디", "붕대", "역서", "미라다의돌")
	RETURNF 1

@PRINT_ITEM_EXPLANATION(ARG)
PRINTFORM [{ARG, 3, RIGHT}] %ITEMNAME:ARG, 18, LEFT% {ITEM:ARG, 3, RIGHT}개
SETCOLOR C_L_GRAY
	PRINTFORML %ITEM_EXPLANATION(ARG)%
RESETCOLOR

@약학지식(ARG)
#FUNCTION
#DIM 회복량
#DIM 강화횟수
회복량 = ARG
;回復量 = 基準値 * (1.1 ^ (조제Lv + 薬に関係しそうな知識))
강화횟수 = TALENT:MASTER:조제Lv + TALENT:MASTER:금단의지식 + TALENT:MASTER:조합지식
WHILE 강화횟수 > 0
	TIMES 회복량, 1.1
	강화횟수 --
WEND
RETURNF 회복량

;미각에 대한 회복 보정
;만복도가 오르지 않는 식품은 기력만 보정
@ITEM_TASTE(CHARA, 기준치, 식품특징)
#FUNCTION
#DIM CHARA
#DIM 기준치
#DIMS 식품특징
#DIMS TASTE_CHARA_GOOD
#DIMS TASTE_CHARA_BAD
#DIM 회복량
TASTE_CHARA_GOOD = %GET_STR(, "캐릭터데이터", CHARA, "요리：좋아하는 맛")%
TASTE_CHARA_BAD = %GET_STR(, "캐릭터데이터", CHARA, "요리：싫어하는 맛")%

회복량 = 기준치 + (STR_MULTI_COUNT(TASTE_CHARA_GOOD, 식품특징) - STR_MULTI_COUNT(TASTE_CHARA_BAD, 식품특징)) * 10

;제철 등 계절에 따른 보정
SELECTCASE GET_MONTH()
	CASE "봄의 달"
		IF STRCOUNT(식품특징, "봄철/") && !STRCOUNT(TASTE_CHARA_BAD, "봄철/")
			TIMES 회복량, 1.3
		ENDIF
	CASE "여름의 달"
		;夏は아이스크림が2.0倍(嫌いじゃなければ）
		IF STRCOUNT(식품특징, "아이스크림/") && !STRCOUNT(TASTE_CHARA_BAD, "아이스크림/")
			TIMES 회복량, 2.0
		ELSEIF STRCOUNT(식품특징, "수박/")
			TIMES 회복량, 1.3
		ENDIF
	CASE "가을의 달"
		IF STRCOUNT(식품특징, "으름덩굴/")
			TIMES 회복량, 1.3
		ENDIF
	CASE "겨울의 달"
		;冬は아이스크림が0.5倍(好きじゃなければ）
		IF STRCOUNT(식품특징, "아이스크림/") && !STRCOUNT(TASTE_CHARA_GOOD, "아이스크림/")
			TIMES 회복량, 0.5
		ENDIF
ENDSELECT

RETURNF 회복량

;아이템
@COM_ABLE490
;移動実行判定
SIF !TFLAG:100
	RETURN 0
;一括管理
SIF GLOBAL_COMABLE(490)
	RETURN RESULT
RETURN 1

path: ERB/커맨드관련/COMF/일상계/COMF490 아이템.ERB


code:
;-------------------------------------------------
;화이트데이
;-------------------------------------------------
@COM491

#DIM CAN_COOK
#DIM 호감도상승량
#DIM 신뢰도상승량
#DIM WD_BUFF
#DIMS WD_GIFT

호감도상승량 = 0
신뢰도상승량 = 0
CAN_COOK = 0
WD_BUFF = 0

;調理可能か
CALL COM_ABLE413
CAN_COOK = RESULT
IF FESTIVAL() == "화이트 데이" && INRANGE(CFLAG:TARGET:발렌타인,1,4) && SHIRAHU(TARGET)
	CALL ASK_M("답례 상대 목록",1,"답례용 과자를 만든다 (감미재 5개)",ITEM:감미재 >= 5,@"%CALLNAME:TARGET%에게 답례품을 전달",1)
ELSE
	CALL ASK_M("답례 상대 목록",1,"답례용 과자를 만든다",1)
ENDIF
SELECTCASE RESULT
	CASE 0
		CALL Valentine_List
		WAIT
		RETURN -1
	CASE 1
		IF !CAN_COOK
			PRINTFORMW 요리를 할 수 있는 환경이 필요하다
			RETURN -1
		ENDIF
		CALL WD_COOK
		RETURN RESULT
	CASE 2
		PRINTFORML 무엇을 건네줄까？（받은 것：%Valentine_Chocolate(TARGET)%）
		{
		CALL ASK_M("그만둔다",1,"블랙썬더",ITEM:블랙썬더,"로터초코",ITEM:로터초코,"WD초콜릿",ITEM:WD초콜릿,
		"수제초코쿠키",ITEM:수제초코쿠키,"WD푸딩",ITEM:WD푸딩,
		"수제푸딩",ITEM:수제푸딩,"데코레이션케이크",ITEM:데코레이션케이크 && CFLAG:TARGET:발렌타인 == 4)
		}
		SELECTCASE RESULT
			CASE 0
				RETURN -1
			CASE 1
				WD_GIFT = 블랙썬더
				SIF CFLAG:발렌타인 > 1
					PRINTFORMW %타겟은% 석연치 않은 얼굴을 하고 있다……
				호감도상승량 = 30
				신뢰도상승량 = 30
			CASE 2
				WD_GIFT = 로터초코
				호감도상승량 = 50
				신뢰도상승량 = 50
				SOURCE:쾌C += 3000
				SOURCE:반감 += 300
			CASE 3
				WD_GIFT = WD초콜릿
				호감도상승량 = 100
				신뢰도상승량 = 100
				WD_BUFF = 100
			CASE 4
				WD_GIFT = 수제초코쿠키
				호감도상승량 = 150
				신뢰도상승량 = 150
				EXP:애정경험 += 1
				WD_BUFF = 100
			CASE 5
				WD_GIFT = WD푸딩
				호감도상승량 = 200
				신뢰도상승량 = 200
				EXP:애정경험 += 3
				WD_BUFF = 100
			CASE 6
				WD_GIFT = 수제푸딩
				호감도상승량 = 300
				신뢰도상승량 = 300
				EXP:애정경험 += 5
				WD_BUFF = 200
			CASE 7
				WD_GIFT = 데코레이션케이크
				호감도상승량 = 1000
				신뢰도상승량 = 1000
				EXP:애정경험 += 30
				WD_BUFF = 1000
		ENDSELECT
		TFLAG:193 = RESULT
		SIF RESULT >= 5
			TALENT:기분 = 1
		PRINTFORMW %타겟에게% %조사처리(WD_GIFT,"를")% 선물했다
		ITEM:WD_GIFT --
		CALL CHANGE_CFLAG(2,TARGET,호감도상승량)
		CALL CHANGE_CFLAG(4,TARGET,신뢰도상승량)
		IF WD_BUFF
			CALL BUFF_BASE(TARGET,BASE_체력, WD_BUFF, 1)
			CALL BUFF_BASE(TARGET,BASE_기력, WD_BUFF, 1)
		ENDIF
		CFLAG:TARGET:발렌타인 += 100
		TIME += 10
		RETURN 1
ENDSELECT

@Valentine_List
FOR LOCAL,1,CHARANUM
	SIF !CFLAG:LOCAL:발렌타인
		CONTINUE
	SELECTCASE CFLAG:LOCAL:발렌타인 % 100
		CASE 2
			SETCOLOR C_GREEN
		CASE 3
			SETCOLOR C_YELLOW
		CASE 4
			SETCOLOR C_PINK
	ENDSELECT
	;返礼済みなら打ち消し線
	SIF CFLAG:LOCAL:발렌타인 > 100
		FONTSTYLE 4
	PRINTFORMLC [{LOCAL}]%CALLNAME:LOCAL%
	FONTREGULAR
	RESETCOLOR
NEXT
PRINTFORML
DRAWLINE
SETCOLOR C_PINK
PRINTFORM 수제
SETCOLOR C_YELLOW
PRINTFORM 고급
SETCOLOR C_GREEN
PRINTFORM 보통
RESETCOLOR
PRINTFORML 의리

@WD_COOK
PRINTFORML 무엇을 만들까？（전부 감미재５개 소비）
{
CALL ASK_M("그만둔다",1, "초코쿠키×10＋α：대량으로 만들 수 있다。흩뿌리기용。",1,
"푸딩×5＋α：밸런스가 좋다。벌꿀이 필요。（필요: 벌꿀）",ITEM:벌꿀,
"데코레이션케이크：소중한 사람에게。（필요: 딸기３・요리기능４）",ITEM:딸기 >= 3 && ABL:MASTER:요리기능 >= 4,
"로터초코×5：로터를 사용한 먹을 수 있는 수수께끼의 초콜릿。（필요: 로터）",ITEM:로터)
}
SELECTCASE RESULT
	CASE 0
		RETURN -1
	CASE 1
		CALL GET_ITEM("수제초코쿠키", 10 + ABL:MASTER:요리기능, 2)
		ITEM:감미재 -= 5
	CASE 2
		CALL GET_ITEM("수제푸딩", 5 + ABL:MASTER:요리기능 / 3, 2)
		ITEM:감미재 -= 5
		ITEM:벌꿀 -= 1
	CASE 3
		ITEM:감미재 -= 5
		ITEM:딸기 -= 3
		CALL GET_ITEM("데코레이션케이크", 1, 2)
	CASE 4
		CALL GET_ITEM("로터초코", 5, 2)
		ITEM:감미재 -= 5
		ITEM:로터 -= 1
ENDSELECT
TIME += 60
EXP:MASTER:요리경험 += 3
DOWNBASE:MASTER:기력 = 300
RETURN 1

;-------------------------------------------------
;実行判定
;-------------------------------------------------
@COM_ABLE491
;チョコレートを贈る実行判定
SIF !TFLAG:100
	RETURN 0
;一括管理
SIF GLOBAL_COMABLE(491)
	RETURN RESULT
;気力0
SIF BASE:MASTER:기력 <= 0
	RETURN 0
;停止中は不可
SIF FLAG:70 == 1
	RETURN 0
;添い寝中
SIF CFLAG:동침중
	RETURN 0
;イタズラ中
SIF CFLAG:MASTER:장난
	RETURN 0
SIF CFLAG:우후후 == 2
	RETURN 0
SIF !PREPARE_WD()
	RETURN 0
IF FESTIVAL() != "화이트 데이"
	SIF !KITCHEN(CFLAG:MASTER:현재위치)
		RETURN 0
ENDIF

RETURN 1

@PREPARE_WD()
#FUNCTION
SIF DAY:2 != 1
	RETURNF 0
SIF DAY:3 > 14
	RETURNF 0
RETURNF 1

;デバッグ用
@WD_DEBUG
ITEM:감미재 = 99
ITEM:딸기 = 99
ITEM:벌꿀 = 99
DAY:2 = 1
DAY:3 = 13
FOR LOCAL,1,CHARANUM
	CFLAG:LOCAL:발렌타인 = 1 + RAND:4
NEXT

path: ERB/커맨드관련/COMF/일상계/COMF491 화이트데이.ERB

license: mit


path: ERB/커맨드관련/COMF/일상계/COMF495 하늘을바라본다.ERB


code:
;-------------------------------------------------
;ラジオをいじる
;日常系コマンド、レベル1
;-------------------------------------------------
@COM496
#DIMS 大雨情報
#DIMS 大雪情報
#DIMS 暴風情報
#DIMS 雷情報
#DIMS 低温情報
#DIMS 雪崩情報
#DIMS 低気圧位置
#DIMS 線状降水帯位置
#DIMS 積乱雲位置
#DIMS 台風勢力
#DIMS 早期注意情報

低気圧位置 = %GET_LOCATION_LOW()%
線状降水帯位置 = %GET_LOCATION_LINER_RAINBAND()%
積乱雲位置 = %GET_LOCATION_CUMULUS()%

大雨情報 = %GET_WEATHER_WARNING_HEAVY_RAIN()%
大雪情報 = %GET_WEATHER_WARNING_HEAVY_SNOWFALL()%
暴風情報 = %GET_WEATHER_WARNING_STORM()%
雷情報 = %GET_WEATHER_WARNING_LIGHTNING()%
低温情報 = %GET_WEATHER_WARNING_COLD()%
雪崩情報 = %GET_WEATHER_WARNING_AVALANCHE()%
早期注意情報 = %GET_WEATHER_ADVISORY_probability_subsequent_WARNING()%

SELECTCASE IS_TYPHOON:1
	CASE 5
		台風勢力 = 猛烈な
	CASE 4
		台風勢力 = 非常に強い
	CASE 3
		台風勢力 = 強い
	CASEELSE
		台風勢力 =
ENDSELECT

[IF_DEBUG]
SELECTCASE TIME:0
	CASE 345 TO 359, 375 TO 389, 405 TO 419, 435 TO 449, 465 TO 479, 525 TO 539, 585 TO 599, 705 TO 719, 735 TO 749, 825 TO 839, 885 TO 899, 945 TO 959, 1005 TO 1019, 1065 TO 1079, 1125 TO 1139, 1185 TO 1199, 1245 TO 1259, 1305 TO 1319, 1365 TO 1379
	;天気予報(15分間)(05:45 06:15 06:45 07:15 07:45 08:45 09:45 11:45 12:15 13:45 14:45 15:45 16:45 17:45 18:45 19:45 20:45 21:45 22:45)
		DEBUGPRINTFORML 現在の放送：天気予報
	CASE 300 TO 329, 360 TO 374, 420 TO 434, 540 TO 554, 600 TO 614, 660 TO 674, 720 TO 734, 780 TO 794, 840 TO 854, 900 TO 914, 1020 TO 1034, 1140 TO 1184, 1200 TO 1214, 1260 TO 1274, 1380 TO 1394, 0 TO 14, 60 TO 74, 120 TO 134, 180 TO 194, 240 TO 254
	;ニュース(原則15分間)(05:00~05:29 06:00 07:00 09:00 10:00 11:00 12:00 13:00 14:00 15:00 17:00 19:00~19:44 20:00 21:00 23:00 24:00 01:00 02:00 03:00 04:00)
		DEBUGPRINTFORML 現在の放送：ニュース
	CASE 960 TO 989
	;気象通報(16:00~16:29)
		DEBUGPRINTFORML 現在の放送：気象通報
	CASEELSE
		DEBUGPRINTFORML 現在の放送：不明
ENDSELECT

DEBUGPRINTFORML %大雨情報%
DEBUGPRINTFORML %大雪情報%
DEBUGPRINTFORML %暴風情報%
DEBUGPRINTFORML %雷情報%
DEBUGPRINTFORML %低温情報%
DEBUGPRINTFORML %雪崩情報%
DEBUGPRINTFORML %早期注意情報%

[ENDIF]

PRINTW ラジオのつまみをいじってみた
PRINTL

PRINTL ┏━━━━━━┓
PRINT ┃電源［
IF 大雨情報 == "大雨特別警報"
	PRINT □□
ELSEIF 大雨情報 == "大雨警報(土砂災害警戒情報)" || 大雪情報 == "大雪特別警報" || 暴風情報 == "暴風雪特別警報" || 暴風情報 == "暴風特別警報"
	IF CONTINUOUS_RAINFALL >= 10
		SETCOLOR C_P_PURPLE
		PRINT ■
		RESETCOLOR
		PRINT □
	ELSE
		SETCOLOR C_P_PURPLE
		PRINT ■■
		RESETCOLOR
	ENDIF
ELSEIF 大雨情報 == "大雨警報" || 大雪情報 == "大雪警報" || 暴風情報 == "暴風雪警報" || 暴風情報 == "暴風警報" || 雷情報 == "雷注意報(竜巻注意情報)" || 雪崩情報 == "なだれ注意報"
	SETCOLOR C_RED
	PRINT ■■
	RESETCOLOR
ELSEIF 大雨情報 == "大雨注意報" || 大雪情報 == "大雪注意報" || 暴風情報 == "風雪注意報" || 暴風情報 == "強風注意報" || 雷情報 == "雷注意報"
	SELECTCASE 早期注意情報
		CASE "警報級の可能性［高］"
			SETCOLOR C_YELLOW
			PRINT ■■
			RESETCOLOR
		CASE "警報級の可能性［中］"
			SETCOLOR C_YELLOW
			PRINT ■
			RESETCOLOR
			PRINT ■
		CASEELSE
			SETCOLOR C_YELLOW
			PRINT ■
			SETCOLOR C_GREEN
			PRINT ■
			RESETCOLOR
	ENDSELECT
ELSE
	SELECTCASE 早期注意情報
		CASE "警報級の可能性［高］"
			PRINT ■■
		CASE "警報級の可能性［中］"
			SETCOLOR C_GREEN
			PRINT ■
			RESETCOLOR
			PRINT ■
		CASEELSE
			SETCOLOR C_GREEN
			PRINT ■■
			RESETCOLOR
	ENDSELECT
ENDIF
PRINTL ］┃
PRINTL ┗━━━━━━┛
;(天候パッチ)雷発生時はノイズが混じる
IF GET_FRONTAL_THUNDERSTORM() == 1 && GET_HEAT_THUNDERSTORM() == 1
	PRINT ……
	FONTBOLD
	PRINT ガリガリッ！
	FONTREGULAR
	PRINT ………
	FONTBOLD
	PRINT ジジジ
	FONTREGULAR
	PRINTL ……
ELSEIF GET_FRONTAL_THUNDERSTORM() == 1 || GET_HEAT_THUNDERSTORM() == 1
	PRINT ……
	FONTBOLD
	PRINT ガリッ
	FONTREGULAR
	PRINTL ………　……
ELSE
	PRINTL ……　………　……
ENDIF

SELECTCASE TIME:0
	CASE 345 TO 359, 375 TO 389, 405 TO 419, 435 TO 449, 465 TO 479, 525 TO 539, 585 TO 599, 705 TO 719, 735 TO 749, 825 TO 839, 885 TO 899, 945 TO 959, 1005 TO 1019, 1065 TO 1079, 1125 TO 1139, 1185 TO 1199, 1245 TO 1259, 1305 TO 1319, 1365 TO 1379
	;天気予報(15分間)(05:45 06:15 06:45 07:15 07:45 08:45 09:45 11:45 12:15 13:45 14:45 15:45 16:45 17:45 18:45 19:45 20:45 21:45 22:45)
		SELECTCASE 低気圧位置
			CASE "台風遠方"
				PRINTFORML …の南海上にある%台風勢力%台風…号は……
			CASE "台風勢力圏"
				PRINTFORML …沖にある%台風勢力%台風……中心気圧は…hPa…
			CASE "台風接近強風域", "台風接近暴風域"
				PRINTFORM …%台風勢力%台風…号は
				IF 暴風情報 == "暴風特別警報"
					PRINTFORML ……{TIME_CLOSEST_TYPHOON:1}日%GET_NAME_WEATHER_TIMEZOME(TIME_CLOSEST_TYPHOON:2)%に…数十年に一度の強さで…に接近…
				ELSEIF IS_TYPHOON:1 >= 3
					PRINTFORML ……{TIME_CLOSEST_TYPHOON:1}日%GET_NAME_WEATHER_TIMEZOME(TIME_CLOSEST_TYPHOON:2)%に…勢力を保ったまま…に接近…
				ELSE
					PRINTFORML ……{TIME_CLOSEST_TYPHOON:1}日%GET_NAME_WEATHER_TIMEZOME(TIME_CLOSEST_TYPHOON:2)%に…に接近…
				ENDIF
			CASE "台風最接近"
				PRINTFORM …%台風勢力%台風…号は
				IF 暴風情報 == "暴風特別警報"
					PRINTFORML ……数十年に一度の強さで…最接近し…厳重な警戒を…
				ELSEIF IS_TYPHOON:1 >= 3
					PRINTFORML ……最接近し…厳重な警戒を…
				ELSE
					PRINTFORML ……最接近し…十分な警戒を…
				ENDIF
			CASE "台風通過暴風域", "台風通過強風域"
				PRINTFORM …%台風勢力%台風…号は
				IF 暴風情報 == "暴風特別警報"
					PRINTFORML ……記録的な強さのまま…通過し…厳重な警戒を…
				ELSEIF IS_TYPHOON:1 >= 3
					PRINTFORML ……勢力を保ったまま…通過し…
				ELSE
					PRINTFORML ……を通過し…
				ENDIF
			CASE "寒冷渦遠方", "寒冷渦接近"
				SELECTCASE DAY:2
					CASE 1, 3, 4
						PRINTFORML …寒冷渦が日本海側に進み……
					CASE 2
						PRINTFORML …上空に強い寒気が流れ込み……
				ENDSELECT
			CASE "二つ玉低気圧遠方", "日本海低気圧遠方"
				PRINTFORML …日本海側を…低気圧が……
			CASE "二つ玉低気圧接近", "疑似好天", "二つ玉低気圧通過中"
				PRINTFORML …日本海側を急速に発達した低気圧が……
				PRINTFORML …山では天気の急変に……
			CASE "日本海低気圧接近", "日本海低気圧通過中"
				PRINTFORML …日本海側を発達した低気圧が……
			CASE "南岸低気圧遠方"
				PRINTFORML …東シナ海を…低気圧が……
			CASE "南岸低気圧接近"
				PRINTFORML …太平洋側を発達した低気圧が……
			CASE "温暖前線遠方"
				IF TIME_ENTER_TERRITORY_EXTRATROPICAL_LOW:0 == DAY:2 && TIME_ENTER_TERRITORY_EXTRATROPICAL_LOW:1 <= DAY:3 + 1
					PRINTFORML …明日以降に前線が……
				ELSE
					PRINTFORML …週の中ごろに低気圧が……
				ENDIF
			CASE "温暖前線接近"
				PRINTFORML …温暖前線の接近により西から……
			CASE "暖域"
				PRINTFORML …寒冷前線通過の際には……
			CASE "寒冷前線通過"
				PRINTFORML …ところによりにわか雨が……
			CASEELSE
				SELECTCASE DAY:2
					CASE 1,3
						IF IS_RAINY_SEASON1:0 == 1 || IS_RAINY_SEASON2 == 1
							PRINTFORML …はぐずついた空模様……
						ELSE
							PRINTFORML …は穏やかな空模様……
						ENDIF
					CASE 2
						IF POWER_WARM_AIR == 1
							PRINTFORML …太平洋高気圧の張り出しが弱く……
						ELSE
							PRINTFORML …全国的に夏空が……
						ENDIF
					CASE 4
						PRINTFORML …西高東低…冬型の気圧配置に……
				ENDSELECT
		ENDSELECT

		SELECTCASE 大雨情報
			CASE "大雨特別警報"
				SELECTCASE RAND:4
					CASE 0
						PRINTFORML ……これまでに経験したことのないような大雨が……
					CASE 1
						PRINTFORML …に数十年に一度の降水量が……
					CASE 2
						PRINTFORML ……地域に…大雨特別警報を……
					CASE 3
						PRINTFORML ……既に安全な避難が……命を守るための行動を……
				ENDSELECT
			CASE "大雨警報(土砂災害警戒情報)"
				SELECTCASE RAND:2
					CASE 0
						PRINTFORML …過去数年間で最も土砂災害の危険性が……
					CASE 1
						PRINTFORML …象台は…地域に土砂災害の危険性が……
				ENDSELECT
			CASE "大雨警報"
				SELECTCASE RAND:3
					CASE 0
						PRINTFORML …などの移動が困難な方は……
					CASE 1
						PRINTFORML …明るいうちに…避難を……
					CASE 2
						PRINTFORML …象台は…地域に大雨・洪水警報を……
				ENDSELECT
			CASE "大雨注意報"
				SELECTCASE RAND:2
					CASE 0
						PRINTFORML …お住いの地域のハザードマップを……
					CASE 1
						PRINTFORML …象台は…地域に大雨注意報を……
				ENDSELECT
		ENDSELECT

		SELECTCASE 大雪情報
			CASE "大雪特別警報"
				SELECTCASE RAND:3
					CASE 0
						PRINTFORML …豪雪以来の大雪となって……
					CASE 1
						PRINTFORML …数十年に一度の大雪が……
					CASE 2
						PRINTFORML …地域に大雪特別警報を……
				ENDSELECT
			CASE "大雪警報"
				IF CONTINUOUS_SNOWFALL >= 20
					PRINTFORML ……象台は…地域に顕著な大雪に関する……
				ELSE
					PRINTFORML ……地域には大雪警報が……
				ENDIF
			CASE "大雪注意報"
				PRINTFORML ……地域には大雪注意報が……
		ENDSELECT

		SELECTCASE 暴風情報
			CASE "暴風雪特別警報"
				SELECTCASE RAND:4
					CASE 0
						PRINTFORML ……地域に…暴風雪特別警報を……
					CASE 1
						PRINTFORML …数十年に一度の猛吹雪が……
					CASE 2
						PRINTFORML …これまでにない猛吹雪が…不要不急の外出は……
					CASE 3
						PRINTFORML …猛烈な吹雪が…いつ発生してもおかしく……
				ENDSELECT
			CASE "暴風特別警報"
				SELECTCASE RAND:4
					CASE 0
						PRINTFORML ……地域に…暴風特別警報を……
					CASE 1
						PRINTFORML …これまでに経験のない猛烈な風が……
					CASE 2
						PRINTFORML …湾台風以来の…記録的な……
					CASE 3
						PRINTFORML …猛烈な風が…身の安全を……
				ENDSELECT
			CASE "暴風雪警報"
				SELECTCASE RAND:2
					CASE 0
						PRINTFORML …地域には暴風雪警報が…外出は控えてくだ……
					CASE 1
						PRINTFORML …数年に一度の猛吹雪が……
				ENDSELECT
			CASE "暴風警報"
				PRINTFORML …地域には暴風警報が…早めの避難を……
			CASE "風雪注意報"
				PRINTFORML ……地域には風雪注意報が……
			CASE "強風注意報"
				PRINTFORML ……地域には強風注意報が……
		ENDSELECT

		SELECTCASE 雷情報
			CASE "雷注意報(竜巻注意情報)"
					PRINTFORML …大気の状態が非常に不安定となって……
					PRINTFORML …地域には竜巻…のおそれが高まって……
			CASE "雷注意報"
				IF DAY:2 == 2 && (IS_COLD_LOW || 低気圧位置 == "台風勢力圏")
					PRINTFORML …大気の状態が非常に不安定となって……
				ELSE
					PRINTFORML …大気の状態が不安定に……
				ENDIF
		ENDSELECT

		SELECTCASE 雪崩情報
			CASE "なだれ注意報"
				PRINTFORML ……では雪崩が発生するおそれが……
		ENDSELECT

		SELECTCASE 低温情報
			CASE "低温注意報"
				PRINTFORML …地域は平年と比べて気温が低く……
		ENDSELECT

	CASE 300 TO 329, 360 TO 374, 420 TO 434, 540 TO 554, 600 TO 614, 660 TO 674, 720 TO 734, 780 TO 794, 840 TO 854, 900 TO 914, 1020 TO 1034, 1140 TO 1184, 1200 TO 1214, 1260 TO 1274, 1380 TO 1394, 0 TO 14, 60 TO 74, 120 TO 134, 180 TO 194, 240 TO 254
	;ニュース(原則15分間)(05:00~05:29 06:00 07:00 09:00 10:00 11:00 12:00 13:00 14:00 15:00 17:00 19:00~19:44 20:00 21:00 23:00 24:00 01:00 02:00 03:00 04:00)
		SELECTCASE 低気圧位置
			CASE "台風遠方"
				PRINTFORML …の南海上にある%台風勢力%台風…号は……
			CASE "台風勢力圏"
				PRINTFORML …沖にある%台風勢力%台風……中心気圧は…hPa…
			CASE "台風接近強風域", "台風接近暴風域", "台風最接近", "台風通過暴風域", "台風通過強風域"
				PRINTFORM …%台風勢力%台風…号は
				IF IS_TYPHOON:1 >= 3
					PRINTFORML ……勢力を保ったまま…に接近…
				ELSE
					PRINTFORML ……日には…に接近…
				ENDIF
			CASE "寒冷渦遠方", "寒冷渦接近"
				SELECTCASE DAY:2
					CASE 1, 3, 4
						PRINTFORML …寒冷渦が日本海側に進み……
					CASE 2
						PRINTFORML …上空に強い寒気が流れ込み……
				ENDSELECT
			CASE "二つ玉低気圧接近", "疑似好天", "二つ玉低気圧通過中"
				PRINTFORML …山での遭難が相次いで……
		ENDSELECT

		SELECTCASE 大雨情報
			CASE "大雨特別警報"
				SELECTCASE RAND:5
					CASE 0
						PRINTFORML ……これまでに経験したことのないような大雨が……
					CASE 1
						PRINTFORML …に数十年に一度の降水量が……
					CASE 2
						PRINTFORML ……地域に…大雨特別警報を……
					CASE 3
						PRINTFORML ……の警戒レベルが5に……
					CASE 4
						PRINTFORML ……既に安全な避難が……命を守るための行動を……
				ENDSELECT

				SELECTCASE CONTINUOUS_RAINFALL
					CASE IS >= 40
						SELECTCASE RAND:2
							CASE 0
								PRINTFORML …政府は…激甚災害に指定……
							CASE 1
								PRINTFORML …各地で…寸断され……が救助……
						ENDSELECT
					CASE IS >= 20
						SELECTCASE RAND:3
							CASE 0
								PRINTFORML …土砂崩れにより…世帯…が孤立……
							CASE 1
								PRINTFORML …既に記録的な大雨になっている地域も……
							CASE 2
								PRINTFORML …川で氾濫危険水位を超え……
						ENDSELECT
					CASE IS >= 10
						SELECTCASE RAND:2
							CASE 0
								PRINTFORML …市に記録的短時間大雨情報を……
							CASE 1
								PRINTFORML …に数年に一度の大雨が……
						ENDSELECT
				ENDSELECT
			CASE "大雨警報(土砂災害警戒情報)"
				SELECTCASE RAND:5
					CASE 0
						PRINTFORML ……の警戒レベルが4に……
					CASE 1
						PRINTFORML …過去数年間で最も土砂災害の危険性が……
					CASE 2
						PRINTFORML …象台は…地域に土砂災害の危険性が……
					CASE 3
						PRINTFORML …市は……世帯…に避難指示を……
					CASE 4
						PRINTFORML …線は全線で運転を見合わせ……
				ENDSELECT

			CASE "大雨警報"
				SELECTCASE RAND:5
					CASE 0
						PRINTFORML …などの移動が困難な方は……
					CASE 1
						PRINTFORML …明るいうちに…避難を……
					CASE 2
						PRINTFORML …象台は…地域に大雨・洪水警報を……
					CASE 3
						PRINTFORML …市は……世帯…に高齢者等避難を……
					CASE 4
						PRINTFORML ……の警戒レベルが2に……
				ENDSELECT
		ENDSELECT

		SELECTCASE 大雪情報
			CASE "大雪特別警報"
				SELECTCASE RAND:4
					CASE 0
						PRINTFORML …豪雪以来の大雪となって……
					CASE 1
						PRINTFORML …数十年に一度の大雪が……
					CASE 2
						PRINTFORML …地域に大雪特別警報を……
					CASE 3
						PRINTFORML …記録的な大雪により各地で…孤立……
				ENDSELECT
			CASE "大雪警報"
				IF CONTINUOUS_SNOWFALL >= 20
					SELECTCASE RAND:2
						CASE 0
							PRINTFORML ……象台は…地域に顕著な大雪に関する……
						CASE 1
							PRINTFORML …大雪により大規模な交通障害が……
					ENDSELECT
				ELSE
					SELECTCASE RAND:2
						CASE 0
							PRINTFORML …象台は…地域に大雪警報を……
						CASE 1
							PRINTFORML …大雪により各地の交通機関に影響が……
					ENDSELECT
				ENDIF
		ENDSELECT

		SELECTCASE 暴風情報
			CASE "暴風雪特別警報"
				SELECTCASE RAND:6
					CASE 0
						PRINTFORML ……地域に…暴風雪特別警報を……
					CASE 1
						PRINTFORML …数十年に一度の猛吹雪が……
					CASE 2
						PRINTFORML …これまでにない猛吹雪が…不要不急の外出は……
					CASE 3
						PRINTFORML …猛烈な吹雪が…いつ発生してもおかしく……
					CASE 4
						PRINTFORML …各地で…が孤立…に救助を要請……
					CASE 5
						PRINTFORML …自動車道では…多くの車が立ち往生……
				ENDSELECT
			CASE "暴風特別警報"
				SELECTCASE RAND:4
					CASE 0
						PRINTFORML ……地域に…暴風特別警報を……
					CASE 1
						PRINTFORML …これまでに経験のない猛烈な風が……
					CASE 2
						PRINTFORML …湾台風以来の…記録的な……
					CASE 3
						PRINTFORML …猛烈な風が…身の安全を……
				ENDSELECT
			CASE "暴風雪警報"
				SELECTCASE RAND:2
					CASE 0
						PRINTFORML …地域には暴風雪警報が…外出は控えてくだ……
					CASE 1
						PRINTFORML …数年に一度の猛吹雪が……
				ENDSELECT
			CASE "暴風警報"
				SELECTCASE RAND:2
					CASE 0
						PRINTFORML …象台は…地域に暴風警報を……
					CASE 1
						PRINTFORML …風が…からの避難は危険…早めの避難を……
				ENDSELECT
		ENDSELECT

		SELECTCASE 雷情報
			CASE "雷注意報(竜巻注意情報)"
				PRINTFORML …大気の状態が非常に不安定となって……
				PRINTFORML …地域には竜巻…のおそれが高まって……
			CASE "雷注意報"
				IF DAY:2 == 2 && (IS_COLD_LOW || 低気圧位置 == "台風勢力圏")
					PRINTFORML …大気の状態が非常に不安定となって……
				ENDIF
		ENDSELECT

		IF POSSIBLITY_AVALANCHE >= 1
			PRINTFORML …各地で雪崩による被害が……
		ENDIF

		SIF IS_RAINY_SEASON1:1 == 1
			PRINTFORML …象台は…西日本に梅雨明け宣言を……

		SELECTCASE IS_COLD_CROP_DAMAGE
			CASE IS >= 3
				PRINTFORML …記録的な冷害により……緊急で……
			CASE IS >= 1
				PRINTFORML …天候不順により作物に影響が……
		ENDSELECT

		SELECTCASE RAND:3
			CASE 0
				PRINTFORML …今日の円相場は……
			CASE 1
				PRINTFORML …続きまして各地のニュースを……
			CASE 2
				PRINTFORML …によりますと……
		ENDSELECT

	CASE 960 TO 989
	;気象通報(16:00~16:29)
		PRINTFORML …発表の…{DAY:3}日正午の気象通報です……
		SELECTCASE RAND:54
			CASE 0
				PRINTFORML ……引き続き…{DAY:3}日正午の天気をお伝え……
			CASE 1
				PRINTFORML …富士山…気温……
			CASEELSE
				{
				PRINTFORML …%TEXTR("/石垣島/那覇/名瀬/鹿児島/福江/厳原/足摺岬/室戸岬/松山/浜田/西郷/大阪/潮岬/
				八丈島/大島/御前崎/銚子/前橋/小名浜/輪島/相川/仙台/宮古/秋田/函館/浦河/根室/稚内/
				ポロナイスク/セベロクリリスク/ハバロフスク/ルドナヤプリスタ/ウラジオストク/
				ソウル/ウルルン島/プサン/モッポ/チェジュ島/台北/恒春/長春/北京/大連/青島/
				上海/武漢/アモイ/香港/バスコ/マニラ/父島/南鳥島/
				")%…風力……hPa……
				}
		ENDSELECT

		PRINTFORML …つづいて漁業気象です

		SELECTCASE 低気圧位置
			CASE "台風遠方", "台風勢力圏", "台風接近強風域", "台風接近暴風域", "台風最接近", "台風通過暴風域", "台風通過強風域"
				PRINTFORML ……には…hPaの台風…号があって…kmで進んで……
			CASE "寒冷渦遠方", "寒冷渦接近"
				PRINTFORML ……には…hPaの低気圧があって…ほとんど停滞……
			CASE "二つ玉低気圧遠方", "日本海低気圧遠方"
				PRINTFORML 黄海の…には…hPaの低気圧があって…kmで進んで……
			CASE "日本海低気圧接近", "日本海低気圧通過中"
				PRINTFORML 日本海の…には…hPaの低気圧があって…kmで進んで……
			CASE "二つ玉低気圧接近", "疑似好天", "二つ玉低気圧通過中"
				PRINTFORML 日本海の…には…hPaの低気圧があって…kmで進んで……
				PRINTFORML 本州の…には…hPaの低気圧があって…kmで進んで……
			CASE "南岸低気圧遠方"
				PRINTFORML 東シナ海の…には…hPaの低気圧があって…kmで進んで……
			CASE "南岸低気圧接近"
				PRINTFORML 本州の…には…hPaの低気圧があって…kmで進んで……
			CASE "温暖前線遠方"
				PRINTFORML 長江下流の…には…hPaの低気圧があって…kmで進んで……
			CASE "温暖前線接近"
				PRINTFORML 西日本の…には…hPaの低気圧があって…kmで進んで……
			CASE "暖域"
				PRINTFORML 北陸の…には…hPaの低気圧があって…kmで進んで……
			CASE "寒冷前線通過"
				PRINTFORML 三陸沖の…には…hPaの低気圧があって…kmで進んで……
			CASE "低気圧未発生"
				PRINTFORML …の…には…hPaの高気圧があって…ほとんど停滞……
		ENDSELECT
	CASEELSE
		IF GET_LOCATION_AREA_TYPHOON() == 1 && IS_TYPHOON:1 >= 5
			SELECTCASE RAND:5
				CASE 0
					IF 暴風情報 == "暴風特別警報"
						PRINTFORML …猛烈な台風…号が記録的な勢力のまま接近……
					ELSE
						PRINTFORML …猛烈な台風…号が接近……
					ENDIF
				CASE 1
					IF 暴風情報 == "暴風特別警報"
						PRINTFORML …台風…号により各地に甚大な被害が……
					ELSE
						PRINTFORML …台風…号により各地に被害が……
					ENDIF
				CASE 2
					PRINTFORML …線は全線で運転を見合わせ……
				CASE 3
					IF 暴風情報 == "暴風特別警報"
						PRINTFORML …湾台風以来の…身の安全を確保……
					ELSE
						PRINTFORML …できるだけ早いうちに避難を……
					ENDIF
				CASE 4
					PRINTFORML …空の便にも大きな影響が……
			ENDSELECT
		ENDIF
		SELECTCASE 大雨情報
			CASE "大雨特別警報"
				SELECTCASE RAND:5
					CASE 0
						PRINTFORML ……これまでに経験したことのないような大雨が……
					CASE 1
						PRINTFORML …に数十年に一度の降水量が……
					CASE 2
						PRINTFORML ……地域に…大雨特別警報を……
					CASE 3
						PRINTFORML ……の警戒レベルが5に……
					CASE 4
						PRINTFORML ……既に安全な避難が……命を守るための行動を……
				ENDSELECT

				SELECTCASE CONTINUOUS_RAINFALL
					CASE IS >= 40
						SELECTCASE RAND:2
							CASE 0
								PRINTFORML …政府は…激甚災害に指定……
							CASE 1
								PRINTFORML …各地で…寸断され……が救助……
						ENDSELECT
					CASE IS >= 20
						SELECTCASE RAND:3
							CASE 0
								PRINTFORML …土砂崩れにより…世帯…が孤立……
							CASE 1
								PRINTFORML …既に記録的な大雨になっている地域も……
							CASE 2
								PRINTFORML …川で氾濫危険水位を超え……
						ENDSELECT
					CASE IS >= 10
						SELECTCASE RAND:2
							CASE 0
								PRINTFORML …市に記録的短時間大雨情報を……
							CASE 1
								PRINTFORML …に数年に一度の大雨が……
						ENDSELECT
				ENDSELECT
			CASE "大雨警報(土砂災害警戒情報)"
				SELECTCASE RAND:5
					CASE 0
						PRINTFORML ……の警戒レベルが4に……
					CASE 1
						PRINTFORML …過去数年間で最も土砂災害の危険性が……
					CASE 2
						PRINTFORML …象台は…地域に土砂災害の危険性が……
					CASE 3
						PRINTFORML …市は……世帯…に避難指示を……
					CASE 4
						PRINTFORML …線は全線で運転を見合わせ……
				ENDSELECT
			CASEELSE
				IF 暴風情報 == "暴風雪特別警報"
					SELECTCASE RAND:6
						CASE 0
							PRINTFORML ……地域に…暴風雪特別警報を……
						CASE 1
							PRINTFORML …数十年に一度の猛吹雪が……
						CASE 2
							PRINTFORML …これまでにない猛吹雪が…不要不急の外出は……
						CASE 3
							PRINTFORML …猛烈な吹雪が…いつ発生してもおかしく……
						CASE 4
							PRINTFORML …各地で…が孤立…に救助を要請……
						CASE 5
							PRINTFORML …自動車道では…多くの車が立ち往生……
					ENDSELECT
				ELSEIF 大雪情報 == "大雪特別警報"
					SELECTCASE RAND:4
						CASE 0
							PRINTFORML …豪雪以来の大雪となって……
						CASE 1
							PRINTFORML …数十年に一度の大雪が……
						CASE 2
							PRINTFORML …地域に大雪特別警報を……
						CASE 3
							PRINTFORML …記録的な大雪により各地で…孤立……
					ENDSELECT
				ELSEIF 大雪情報 == "大雪警報" && CONTINUOUS_SNOWFALL >= 20
					SELECTCASE RAND:2
						CASE 0
							PRINTFORML ……象台は…地域に顕著な大雪に関する……
						CASE 1
							PRINTFORML …大雪により大規模な交通障害が……
					ENDSELECT
				ELSEIF 雷情報 == "雷注意報(竜巻注意情報)"
					SELECTCASE RAND:3
						CASE 0
							PRINTFORML …地域には竜巻…のおそれが高まって……
						CASE 1
							PRINTFORML …積乱雲が近づいて…安全な場所へ……
						CASE 2
							PRINTFORML …各地で突風や雹による被害が……
					ENDSELECT
				ELSEIF POSSIBLITY_AVALANCHE >= 1
					PRINTFORML …各地で雪崩による被害が……
				ELSE
					PRINTFORML … ……………… ……
				ENDIF
		ENDSELECT
ENDSELECT

TIME += 15

;-------------------------------------------------
;実行判定
;-------------------------------------------------
@COM_ABLE496
SIF !TFLAG:100
	RETURN 0
;停止中は不可
SIF FLAG:70
	RETURN 0
;ウフフ中は不可
SIF CFLAG:우후후
	RETURN 0
;睡眠イタズラ中は不可
SIF CFLAG:장난
	RETURN 0
;添い寝中
SIF CFLAG:동침중
	RETURN 0
;デート中
;SIF CHK_DATENOW(CFLAG:デート中) && CFLAG:現在位置 != OMANEKIBEYA()
;	RETURN 0

RETURN 1

path: ERB/커맨드관련/COMF/일상계/COMF496 라디오를 만진다.ERB


code:
@CRITTER_DATA(C_ID)
#DIM C_ID

;CritterName = 獲物の名前
;CritterMeat = 入手できる肉のアイテム番号 701;ヤマドリ 702;獣肉 703;鳥肉 704;牛肉(未実装) 705;得体の知れない肉
;CritterMeat_Value = 入手できる肉の量（基礎）
;CritterEffort = 獲物の解体時間　TSPの消費量
;CritterIntro = 説明文
;CritterHint = ヒント文
;CritterTribe = 種族・特徴のタグ
;				獣/鳥/ 罠の対象判定に   小型/中型/大型/謎/ 捕獲時の歓楽等補正判定に
;				HUNTING_SPOT対応	博=1 竹=2 魔=3 沢=4 山=5 底=6
;				希少/猛禽/生体/外来/毛皮/水棲/絶滅/地中/爬虫類/水鳥/山鳥/ 今のところフレーバー
;				冬眠/夏鳥/冬鳥/ 季節出現判定に 今のところフレーバー

SELECTCASE C_ID
	CASE 1
		CritterName = 쥐
		CritterMeat = 702;獣肉
		CritterMeat_Value = 1
		CritterEffort = 15
		CritterIntro = 소형의 설치류. 농경민에게는 천적으로, 피해를 막기 위해 다양한 수단을 동원하여 싸워왔다
		CritterHint = 모든 환경에 분포／소형 동물
		CritterTribe = 짐승/소형/하쿠레이/죽림/마법의숲/산기슭/산정/지저/
	CASE 2
		CritterName = 족제비
		CritterMeat = 702
		CritterMeat_Value = 1
		CritterEffort = 30
		CritterIntro = 개아목 족제비과의 소형 짐승. 몸집이 작으면서도 매우 난폭하며, 자기보다 큰 사냥감도 잡아먹는다
		CritterHint = 극단적으로 어두컴컴한 환경 이외에 분포／소형 동물
		CritterTribe = 짐승/소형/하쿠레이/죽림/마법의숲/산기슭/산정/
	CASE 3
		CritterName = 담비
		CritterMeat = 702
		CritterMeat_Value = 1
		CritterEffort = 30
		CritterIntro = 개아목 족제비과의 소형 짐승. 여우는 일곱 번 변신, 너구리는 여덟 번 변신, 담비는 아홉 번 변신, 이라고도 불리는 변화의 달인이라고 하지만…
		CritterHint = 극단적으로 어두컴컴한 환경 이외에 분포／소형 동물
		CritterTribe = 짐승/소형/하쿠레이/죽림/마법의숲/산기슭/산정/모피/
	CASE 4
		CritterName = 밍크
		CritterMeat = 702
		CritterMeat_Value = 1
		CritterEffort = 30
		CritterIntro = 개아목 족제비과의 소형 짐승. 모피가 목적으로 도입된 외래종으로 수달로 종종 오인된다
		CritterHint = 극단적으로 어두컴컴한 환경 이외에 분포／소형 동물
		CritterTribe = 짐승/소형/왜래/하쿠레이/죽림/마법의숲/산기슭/산정/모피/
	CASE 5
		CritterName = 다람쥐
		CritterMeat = 702
		CritterMeat_Value = 1
		CritterEffort = 30
		CritterIntro = 소형의 설치류. 물가에 사는 것이나 뭍에 사는 것, 날아다니는 것 등 종류가 풍부
		CritterHint = 죽림과 극단적으로 어두컴컴한 환경 이외에 분포。동면한다／소형 동물
		CritterTribe = 짐승/소형/동면/하쿠레이/마법의숲/산기슭/산정/
	CASE 6
		CritterName = 라쿤
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 개아목 너구리과의 중형 짐승. 아메리카 대륙 원산의 외래종. 너구리로 오인받는 검은 옆섬 꼬리는 사실 본종의 특징
		CritterHint = 고산과 극단적으로 어두컴컴한 환경 이외에 분포／중형 동물
		CritterTribe = 짐승/중형/왜래/하쿠레이/죽림/마법의숲/산기슭/
	CASE 7
		CritterName = 너구리
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 개아목 개과의 중형 짐승. 고기는 냄새가 심하므로 사전 손질을 잘 할 필요가 있다
		CritterHint = 고산과 극단적으로 어두컴컴한 환경 이외에 분포／중형 동물
		CritterTribe = 짐승/중형/하쿠레이/죽림/마법의숲/산기슭/
	CASE 8
		CritterName = 오소리
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 개아목 족제비과의 중형 짐승. 옛날에는 무지나라는 이름으로 불렸으며, 판 구멍을 너구리 등에게 이용당하는 일이 있다
		CritterHint = 고산 이외에 분포。동면한다／중형 동물
		CritterTribe = 짐승/중형/동면/하쿠레이/죽림/마법의숲/산기슭/지저/
	CASE 9
		CritterName = 흰코사향고양이
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 고양이아목 사향고양이과의 중형 짐승. 외래종으로 생각되고 있지만, 언제쯤 일본에 왔는지는 확실치 않다
		CritterHint = 고산과 극단적으로 어두컴컴한 환경 이외에 분포／중형 동물
		CritterTribe = 짐승/중형/왜래/하쿠레이/죽림/마법의숲/산기슭/모피/
	CASE 10
		CritterName = 여우
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 개아목 개과의 중형 짐승. 개과에는 드물게 떼지어 다니지 않고 작은 가족 단위로 생활한다
		CritterHint = 극단적으로 어두컴컴한 환경 이외에 분포／중형 동물
		CritterTribe = 짐승/중형/하쿠레이/죽림/마법의숲/산기슭/산정/모피/
	CASE 11
		CritterName = 들개
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 야생화된 개. 산야에서 자활하고 있는 것을 노이누(ノイヌ), 일시적 인간으로부터 떨어져 생활하고 있는 것을 노라이누(ノライヌ)라고 한다
		CritterHint = 인계에 가까운 환경을 좋아한다／중형 동물
		CritterTribe = 짐승/중형/하쿠레이/죽림/마법의숲/
	CASE 12
		CritterName = 들고양이
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 야생화한 고양이. 그 포식력은 매우 높아, 작은 동물을 모조리 사냥하는 등 각지에서 문제가 되고 있다
		CritterHint = 인계에 가까운 환경을 좋아한다／중형 동물
		CritterTribe = 짐승/중형/하쿠레이/죽림/마법의숲/
	CASE 13
		CritterName = 멧토끼
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro = 토끼목 토끼과의 소형 짐승. 옛부터 사냥의 대상이었으며, 깔끔한 맛의 토끼 전골 등으로 중용되었다
		CritterHint = 극단적으로 어두컴컴한 환경 이외에 분포　사정이 있어 죽림에 많다／중형 동물
		CritterTribe = 짐승/중형/하쿠레이/죽림/마법의숲/산기슭/산정/모피/
	CASE 14
		CritterName = 뉴트리아
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 45
		CritterIntro= 중형 설치류. 모피 목적으로 도입된 남아메리카 원산의 외래종. 강을 헤엄치는 등 물가에서의 생활을 좋아한다
		CritterHint = 늪이나 강 등 물길을 좋아한다／중형 동물
		CritterTribe = 짐승/중형/수서/왜래/마법의숲/산기슭/모피/
	CASE 15
		CritterName = 멧돼지
		CritterMeat = 702
		CritterMeat_Value = 5
		CritterEffort = 100
		CritterIntro = 경우제목 멧돼지과의 짐승. 상당히 신경질적이며 인간과 생활권이 겹칠 경우 야행성을 나타낸다
		CritterHint = 극단적으로 어두컴컴한 환경 이외에 분포。겨울에 가장 살이 오른다／대형 동물
		CritterTribe = 짐승/대형/하쿠레이/죽림/마법의숲/산기슭/산정/
	CASE 16
		CritterName = 사슴
		CritterMeat = 702
		CritterMeat_Value = 6
		CritterEffort = 100
		CritterIntro = 우제목 사슴과의 짐승. 예로부터 사냥의 대상이나 영수로도 취급된 친숙한 짐승
		CritterHint = 죽림과 극단적으로 어두컴컴한 환경 이외에 분포。가을에 가장 살이 오른다／대형 동물
		CritterTribe = 짐승/대형/하쿠레이/마법의숲/산기슭/산정/모피/
	CASE 17
		CritterName = 영양
		CritterMeat = 702
		CritterMeat_Value = 6
		CritterEffort = 100
		CritterIntro = 우제목 소과 양아과. 현대 일본에서는 수렵이 금지되어 있지만, 과거에는「아오지시(アオジシ)」라고도 불리며 사냥 대상으로 인기였다
		CritterHint = 죽림과 극단적으로 어두컴컴한 환경 이외에 분포。가을에 가장 살이 오른다／대형 동물
		CritterTribe = 짐승/대형/하쿠레이/마법의숲/산기슭/산정/모피/
	CASE 18
		CritterName = 원숭이
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 60
		CritterIntro = 영장목 긴꼬리원숭이과의 짐승. 현대 일본에서는 수렵이 금지되어 있지만, 이전에는「스네(スネ)」라고도 불리며 전문 사냥꾼조차 존재했다
		CritterHint = 인계와 떨어진 환경을 좋아한다／중형 동물
		CritterTribe = 짐승/중형/죽림/산기슭/산정/
	CASE 19
		CritterName = 반달가슴곰
		CritterMeat = 702
		CritterMeat_Value = 7
		CritterEffort = 120
		CritterIntro = 식육목 곰과의 대형 짐승. 곰고기는 물론이고 약이 되는 웅담을 찾아 예로부터 사냥의 대상이 되어 온 짐승
		CritterHint = 인계와 떨어진 환경을 좋아한다。동면한다。가을에 가장 살이 오른다／대형 동물
		CritterTribe = 짐승/대형/동면/마법의숲/산기슭/산정/모피/
	CASE 20
		CritterName = 뱀
		CritterMeat = 705
		CritterMeat_Value = 1
		CritterEffort = 30
		CritterIntro = 파충강 유린목 뱀아목. 껍질은 길한 물건으로, 고기는 잔가시가 많지만 스프 등으로 만들어 먹을 수 있다
		CritterHint = 고산 이외에 분포。동면한다／소형 동물
		CritterTribe = 짐승/소형/파충류/동면/하쿠레이/죽림/마법의숲/산기슭/지저/
	CASE 21
		CritterName = 큰박쥐
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 60
		CritterIntro = 익수목의 포유류. 큰박쥐의 근연종은 과실 등을 먹는 것이 많고 흡혈을 하는 종은 극히 일부분에 한한다
		CritterHint = 어두컴컴한 환경을 좋아한다。동면한다／중형 동물
		CritterTribe = 짐승/중형/동면/마법의숲/지저/
	CASE 22
		CritterName = 두더지
		CritterMeat = 702
		CritterMeat_Value = 1
		CritterEffort = 60
		CritterIntro = 진무맹장목 두더지과의 포유류. 포획에는 땅속에 덫을 둘 필요가 있을 텐데…
		CritterHint = 어두컴컴한 환경을 좋아한다／소형 동물
		CritterTribe = 짐승/소형/땅속/하쿠레이/죽림/마법의숲/산기슭/지저/
	CASE 23
		CritterName = 데스웜
		CritterMeat = 705
		CritterMeat_Value = 15
		CritterEffort = 200
		CritterIntro = 몸길이는 약 50cm이고 성충은 1.5m나 되는 거대한 지렁이나 애벌레 같은 생물. 독을 토한다고도 한다는데…
		CritterHint = 희소종。땅속에 숨어있는 것 같다／대형 동물
		CritterTribe = 짐승/희소/대형/땅속/수수께끼/지저/
	CASE 24
		CritterName = 츠치노코
		CritterMeat = 705
		CritterMeat_Value = 1
		CritterEffort = 60
		CritterIntro = 또 다른 이름은 벌뱀(バチヘビ). 짧고 몸통이 굵은 절구통 형태의 뱀으로, 2m 가까운 도약력을 갖고 맹독을 가진 것으로 알려진 환상의 생물
		CritterHint = 희소종。비교적 밝은 임야에 숨어 있는 것 같다。동면한다／중형 동물
		CritterTribe = 짐승/희소/중형/파충류/동면/하쿠레이/죽림/
	CASE 25
		CritterName = 슬라임
		CritterMeat = 705
		CritterMeat_Value = 5
		CritterEffort = 10
		CritterIntro = 아메바 형태의 점체로 외부로부터 받아드린 것을 소화·분해한다. 생태계에서는 분해자격의 역할을 하고 있다고 하는데…
		CritterHint = ……짐승？　어두컴컴한 환경을 좋아하는 것 같다／중형 동물
		CritterTribe = 짐승/중형/수수께끼/지저/
	CASE 26
		CritterName = 마탕고
		CritterMeat = 604
		CritterMeat_Value = 5
		CritterEffort = 30
		CritterIntro = 왠지 덫에 걸린 사람형으로도 보이는 섬뜩한 분위기의 대형 버섯. 옆에는「버섯은 먹지 마라」라는 메모가 떨어져 있었지만…
		CritterHint = 희소종。버섯이 자라는 환경에 나타나는 것 같다／대형 동물
		CritterTribe = 짐승/희소/대형/수수께끼/마법의숲/
	CASE 27
		CritterName = 야베오오츠노지카
		CritterMeat = 702
		CritterMeat_Value = 8
		CritterEffort = 120
		CritterIntro = 우제목 사슴과. 어깨 높이 1.8m나 되는, 체격도 뿔도 거대한 일본 고유의 사슴. 죠몬 시대 무렵에 멸종했다고 하지만…
		CritterHint = 희소종。신성한 숲의 영수로서 나타난다고 전해진다／대형 동물
		CritterTribe = 짐승/희소/대형/멸종/하쿠레이/
	CASE 28
		CritterName = 하나이즈미모리우시
		CritterMeat = 702
		CritterMeat_Value = 8
		CritterEffort = 120
		CritterIntro = 우제목 소과. 발굴된 유적과 발견자 이름의 한 글자를 따서 하나이즈미(花泉) 모리우(盛牛)라고 명명된, 죠몬 시대에 멸종된 바이슨의 조상
		CritterHint = 희소종。고산에서 조용히 살고 있다고 전해진다／대형 동물
		CritterTribe = 짐승/희소/대형/멸종/산정/
	CASE 29
		CritterName = 일본늑대
		CritterMeat = 702
		CritterMeat_Value = 2
		CritterEffort = 60
		CritterIntro = 개아목 개과의 중형 짐승. 메이지 시대에 멸종한 것으로 알려져 있으며, 예로부터 사람들에게 경외받고, 신으로 취급받았던 짐승
		CritterHint = 희소종。세력권의 산림을 배회하고 있다고 전해진다／중형 동물
		CritterTribe = 짐승/희소/중형/멸종/하쿠레이/죽림/산기슭/
	CASE 30
		CritterName = 일본수달
		CritterMeat = 702
		CritterMeat_Value = 1
		CritterEffort = 45
		CritterIntro = 개아목 족제비과의 소형 동물. 남획과 환경 악화로, 쇼와 말기 이후 목격 사례가 없어, 멸종 취급을 받았으나…
		CritterHint = 희소종。맑은 물 옆에서 조용히 지내고 있다고 전해진다／소형 동물
		CritterTribe = 짐승/희소/소형/수서/멸종/산기슭/
	CASE 31
		CritterName = 로터
		CritterMeat = 0
		CritterMeat_Value = 1
		CritterEffort = 15
		CritterIntro = 덫에 걸려 있던 진동하는 무언가。벌레의 일종이라고도 물고기의 일종이라고도 농작물이나 성인용품이라고도 한다
		CritterHint = ……생물？　온갖 환경에 분포한다／소형 동물
		CritterTribe = 짐승/소형/수서/수수께끼/하쿠레이/죽림/마법의숲/산기슭/산정/

	;ここからは鳥シリーズ
	CASE 50
		CritterName = 야생닭
		CritterMeat = 703
		CritterMeat_Value = 2
		CritterEffort = 30
		CritterIntro = 꿩과 닭속. 아마도 도망쳐 야생화된 닭. 소위 야생종의 닭은 아닌 것 같다
		CritterHint = 인간 마을 근처 임야에 산재／텃새
		CritterTribe = 새/중형/산새/하쿠레이/죽림/마법의숲/
	CASE 51
		CritterName = 뇌조
		CritterMeat = 703
		CritterMeat_Value = 1
		CritterEffort = 30
		CritterIntro = 꿩과 뇌조속. 랭지에 적응해, 고산대에만 분포. 털갈이해 겨울은 새하얀 모습으로 변한다
		CritterHint = 고산에 산재／텃새
		CritterTribe = 새/소형/산새/산정/
	CASE 52
		CritterName = 메추라기
		CritterMeat = 703
		CritterMeat_Value = 1
		CritterEffort = 20
		CritterIntro = 꿩과 메추라기속. 저지대의 초원・농경지 등에서 서식. 꿩과 중 유일하게「이주」를 하는 철새이기도 하다
		CritterHint = 저지대의 밝은 임야에 분포／여름・겨울새
		CritterTribe = 새/소형/산새/여름새/겨울새/하쿠레이/죽림/산기슭/
	CASE 53
		CritterName = 구리꿩
		CritterMeat = 701
		CritterMeat_Value = 3
		CritterEffort = 60
		CritterIntro = 꿩과 구리꿩속. 산간의 삼림에 서식. 전체 길이는 100cm가 넘으며, 수컷은 적갈색 깃털과 검은색・흰색이 섞인 선명한 꼬리깃을 지닌다
		CritterHint = 저지대의 산림에 분포／텃새
		CritterTribe = 새/중형/산새/하쿠레이/마법의숲/산기슭/
	CASE 54
		CritterName = 일본꿩
		CritterMeat = 703
		CritterMeat_Value = 3
		CritterEffort = 60
		CritterIntro = 꿩과 구리꿩속. 산지～평지의 숲, 하천 부지의 초원 등에 서식. 예로부터 사냥의 대상이 되어, 각종 노래와 민화에 등장하는 뛰어난 지명도를 자랑한다
		CritterHint = 저지대의 밝은 임야에 분포／텃새
		CritterTribe = 새/중형/산새/하쿠레이/죽림/마법의숲/산기슭/

ENDSELECT


code:
@DANMAKU_OPPONENT_STATUS(ARG, ARG:1)
;特殊능력표시
LOCAL:1 = 0
FOR LOCAL, 0, 10
	SIF GETBIT(CFLAG:ARG:탄막특능, LOCAL)
		LOCAL:1 ++
NEXT
IF LOCAL:1 == 10
	CALL DANMAKU_SA_OUTPUT(ARG, LOCAL:1)
ELSEIF LOCAL:1 == 0
	PRINTFORML 탄막스킬 미취득
ELSE
	FOR LOCAL, 0, 10
		CALL DANMAKU_SA_OUTPUT(ARG, LOCAL)
	NEXT
ENDIF

SIF ARG:1
	RETURN
;전투능력の差でメッセージ分岐
SELECTCASE ABL:ARG:전투능력 - ABL:MASTER:전투능력
	CASE -6 TO -4
		PRINTFORML 눈 가리고 앉아있어도 이긴다.
	CASE -4 TO -2
		PRINTFORML 눈 가리고 있어도 이길 것 같다.
	CASE -1
		PRINTFORML 질 것 같지는 않다.
	CASE 0
		PRINTFORML 아마 이길 것 같다.
	CASE 1
		PRINTFORML 못 이길 상대는 아니다.
	CASE 2
		PRINTFORML 상대는 꽤 강한 것 같다.
	CASE 3
		PRINTFORML 적어도 당신보다 배는 강한 것 같다.
	CASE 4
		PRINTFORML 기적이 일어나지 않으면 살해당할 것이다.
	CASE 5,6
		PRINTFORML 상대가 거인이라 치면, 당신은 개미똥 이하다.
	CASEELSE

ENDSELECT

@DANMAKU_SA_OUTPUT(OPPONENT, ARG:0)
#DIM OPPONENT

SIF ARG:0 < 10
	LOCAL = GETBIT(CFLAG:OPPONENT:탄막특능, ARG:0)
	IF ARG:0 == 10
		SETCOLOR C_RED
		PRINTFORML 【백재무궁】: 모든 특수 능력을 겸비한다. 최강
	ELSEIF LOCAL && ARG:0 == 0
		SETCOLOR C_P_BLUE
		PRINTFORML 【언소자약】: %CALLNAME:MASTER%의 주사위수 Down
	ELSEIF LOCAL && ARG:0 == 1
		SETCOLOR C_P_BLUE
		PRINTFORML 【온유돈독】: %CALLNAME:MASTER%의 주사위면 Down
	ELSEIF LOCAL && ARG:0 == 2
		SETCOLOR C_RED
		PRINTFORML 【용기늠름】: %CALLNAME:OPPONENT%의 주사위수 Up
	ELSEIF LOCAL && ARG:0 == 3
		SETCOLOR C_RED
		PRINTFORML 【호방뇌락】: %CALLNAME:OPPONENT%의 주사위면 Up
	ELSEIF LOCAL && ARG:0 == 4
		SETCOLOR C_P_PURPLE
		PRINTFORML 【괴력난신】: %CALLNAME:OPPONENT%의 주사위값이 %CALLNAME:MASTER%의 주사위값보다 작아도 그 차이가 일정수 이하라면 잔기가 줄지 않는다
	ELSEIF LOCAL && ARG:0 == 5
		SETCOLOR C_P_PURPLE
		PRINTFORML 【습람심식】: %CALLNAME:OPPONENT%의 초기 잔기가 4가 된다
	ELSEIF LOCAL && ARG:0 == 6
		SETCOLOR C_P_PURPLE
		PRINTFORML 【잔인무도】: %CALLNAME:MASTER%의 피탄에서 복귀했을 때의 폭탄수가 ―1 된다
	ELSEIF LOCAL && ARG:0 == 7
		SETCOLOR C_P_PURPLE
		PRINTFORML 【건곤일척】: %CALLNAME:OPPONENT%가 일정 확률로 폭탄을 사용하게 된다
	ELSEIF LOCAL && ARG:0 == 8
		SETCOLOR C_RED
		PRINTFORML 【견개고고】: %CALLNAME:OPPONENT%의 주사위 수정치 Up
	ELSEIF LOCAL && ARG:0 == 9
		SETCOLOR C_P_BLUE
		PRINTFORML 【유수암한】: %CALLNAME:MASTER%의 주사위 수정치 Down
	ENDIF
RESETCOLOR

;-------------------------------------------------
;戦闘中の処理
;-------------------------------------------------
@DANMAKU_BATTLE(OPPONENT, ENEMY = -1)
#DIM 플레이어잔기
#DIM 상대잔기
#DIM 남은HP
#DIM 남은폭탄횟수
#DIM 턴수
#DIM 플레이어주사위면
#DIM 상대주사위면
#DIM 플레이어주사위수
#DIM 상대주사위수
#DIM 플레이어주사위값
#DIM 상대주사위값
#DIM 플레이어수정치
#DIM 상대수정치
#DIM 상대세기
#DIM 영력충전플래그
#DIM 자만심플래그
#DIM OPPONENT
#DIM ENEMY
#DIM BombPower
#DIM B_Min
#DIM B_Max
#DIM 지참아이템
#DIM DamageThreshold
#DIM Pichun
#DIM Charge
플레이어잔기 = 3
상대잔기 = 3
상대세기 = ABL:OPPONENT:전투능력
남은HP = MAXBASE:MASTER:체력
남은폭탄횟수 = 2
턴수 = 1
영력충전플래그 = 0
자만심플래그 = 0
Pichun = 0
Charge = 0
B_min = ABL:MASTER:전투능력 * 7 * (ABL:MASTER:전투능력 / 2 + 1)
B_Max = ABL:MASTER:전투능력 * 11 * (ABL:MASTER:전투능력 / 2 + 1) + 6
SIF TALENT:OPPONENT:자존심 > 0 && ABL:MASTER:전투능력 < 상대세기 && 상대세기 > 2 && TALENT:OPPONENT:스침 == 1
	자만심플래그 = 1
;【洽覧深識】の효과
SIF GETBIT(CFLAG:OPPONENT:탄막특능, 5)
	상대잔기 ++
;【익스텐드】の효과
SIF TALENT:MASTER:익스텐드 == 1
	플레이어잔기 ++
;【루나슈터】の효과
SIF TALENT:MASTER:루나슈터 == 1
	남은폭탄횟수 ++
;탄막승부補助아이템の持ち込み
CALL DANMAKU_ITEM_USE
지참아이템 = RESULT

;持ち込みアイテム 命のカード(856) と スペルのカード(857) はここで処理
SIF 지참아이템 == 856
	플레이어잔기 ++
SIF 지참아이템 == 857
	남은폭탄횟수 ++

;プレイヤーのダイス数は2.5n + 期待値2 + 期待値1
;ダイス面は(n+1)*期待値10*期待値7*n/2

WHILE 플레이어잔기 > 0 && 상대잔기 > 0
	;주사위수計算
	플레이어주사위수 = (ABL:MASTER:전투능력 + RAND:5) * 2 + (ABL:MASTER:전투능력 + RAND:3) / 2
	상대주사위수 = (상대세기 + RAND:4 + 1) * 2 + 상대세기 / 2 * RAND:3
	SELECTCASE 상대잔기
		CASE IS > 2
			IF 자만심플래그 && 상대잔기 > 2
				CALL COLORMESSAGE(@"%조사처리(CALLNAME:OPPONENT,"는")% %마스터를% 얕잡아 보고 있다",C_YELLOW,1)
				상대주사위수 -= 상대세기 / 2
			ENDIF
		CASE 2
			상대주사위수 += 상대세기 / 2
		CASE 1
			상대주사위수 += 상대세기 / 2 + 1
	ENDSELECT

	;주사위면計算
	플레이어주사위면 = (ABL:MASTER:전투능력 + 1) * (6 + RAND:8) + RAND:(15 + TFLAG:행운보정) * ABL:MASTER:전투능력 / 2
	상대주사위면 = (상대세기 + 1) * (9 + RAND:7) + RAND:10 - (상대잔기 - 1) * 상대세기
	;수정치計算
	IF (상대세기 - ABL:MASTER:전투능력) < 0
		플레이어수정치 = (ABL:MASTER:전투능력 - 상대세기 + 3) * (-RAND:4 - 8) * ABL:MASTER:전투능력 / 2
		상대수정치 = (ABL:MASTER:전투능력 - 상대세기 + 3) * (RAND:4 + 8) * 상대세기 / 2
	ELSEIF (상대세기 - ABL:MASTER:전투능력) > 0
		플레이어수정치 = (상대세기 - ABL:MASTER:전투능력 + 3) * (RAND:4 + 8) * ABL:MASTER:전투능력 / 2
		상대수정치 = (상대세기 - ABL:MASTER:전투능력 + 3) * (-RAND:4 - 8) * 상대세기 / 2
	ELSE
		플레이어수정치 = 0
		상대수정치 = 0
	ENDIF

	;イベント用
	플레이어수정치 -= (HANDICAP_FIXED + TFLAG:GipsLV)
	플레이어수정치 -= RAND:(HANDICAP_RAND + TFLAG:GipsLV * ABL:MASTER:전투능력 + 1)
	;【言笑自若】の효과
	SIF GETBIT(CFLAG:OPPONENT:탄막특능, 0)
		플레이어주사위수 -= (상대세기 / 2 + RAND:3 + 1) * RAND:2
	;【温柔敦厚】の효과
	SIF GETBIT(CFLAG:OPPONENT:탄막특능, 1)
		플레이어주사위면 -= (상대세기 / 2 + 1) * RAND:4 + RAND:5
	;【勇気凛々】の효과
	SIF GETBIT(CFLAG:OPPONENT:탄막특능, 2)
		상대주사위수 += (상대세기 / 2 + RAND:3 + 2) * RAND:2
	;【豪放磊落】の효과
	SIF GETBIT(CFLAG:OPPONENT:탄막특능, 3)
		상대주사위면 += 상대세기 * 2 + 6 / (RAND:3 + 1)
	;【狷介孤高】の효과
	SIF GETBIT(CFLAG:OPPONENT:탄막특능, 8)
		상대수정치 += 상대세기 * (RAND:5 + 5) + RAND:5
	;【幽愁暗恨】の효과
	SIF GETBIT(CFLAG:OPPONENT:탄막특능, 9)
		플레이어수정치 -= 상대세기 * (RAND:4 + 4) + RAND:9

	;【영력충전】の효과
	IF 영력충전플래그
		Charge = 상대세기 / 2 + 1
		플레이어주사위수 += Charge
		CALL COLORMESSAGE(@"【영력충전】에 의해、주사위수가 {Charge}업！",C_RED,1)
		Charge = (상대세기 / 2 + 1) * 2
		플레이어주사위면 += Charge
		CALL COLORMESSAGE(@"【영력충전】에 의해、주사위면이 {Charge}업！",C_RED,1)
		Charge = 상대세기 * (RAND:3 + 2)
		플레이어수정치 += Charge
		CALL COLORMESSAGE(@"【영력충전】에 의해、수정치가 {Charge}업！",C_RED,1)
		영력충전플래그 = 0
	ENDIF

	;【국사무쌍약】使用中　남은HP増大・주사위면に補正
	SIF TCVAR:MASTER:국사무쌍약
		플레이어주사위면 += MAX(RAND:(MAX((TCVAR:MASTER:국사무쌍약), 1) * 2), 1)

	;戦闘バフの効果（プレイヤーと対戦相手両方）
	SIF TCVAR:MASTER:인챈트
		플레이어수정치 += ABL:MASTER:전투능력 * RAND:3 + RAND:3
	SIF TCVAR:OPPONENT:인챈트
		상대수정치 += 상대세기 * RAND:3 + RAND:3
	SIF TCVAR:MASTER:가호
		상대수정치 -= ABL:MASTER:전투능력 * RAND:3 + RAND:3
	SIF TCVAR:OPPONENT:가호
		플레이어수정치 -= 상대세기 * RAND:3 + RAND:3

	;탄막승부補助아이템の효과
	CALL DANMAKU_ITEM_EFFECT(지참아이템)
	플레이어주사위면 += RESULT:0
	플레이어주사위수 += RESULT:1
	;念のためリセット
	VARSET RESULT
	플레이어주사위수 = MAX(플레이어주사위수, 1)
	상대주사위수 = MAX(상대주사위수, 1)
	플레이어주사위면 = MAX(플레이어주사위면, 1)
	상대주사위면 = MAX(상대주사위면, 1)

	PRINTL
	PRINTFORML -----------------------------------------------------
	PRINTFORML TURN:{턴수}
	PRINTFORML -----------------------------------------------------
	PRINTL

	CALL DICE_PRINT(플레이어주사위수, 플레이어주사위면, 플레이어수정치, 상대주사위수, 상대주사위면, 상대수정치, "確定前")
	PRINTFORMW 운명의 주사위 롤!!
	;イベント用
	IF (HANDICAP_FIXED || HANDICAP_RAND)
		CALL KOJO_MESSAGE_SEND("DANMAKU", 상대잔기, OPPONENT, ENEMY, -1, "ハンデ")
	ENDIF
	;ダイス値計算
	플레이어주사위값 = DICE_VALUE(플레이어주사위수, 플레이어주사위면, 플레이어수정치)
	상대주사위값 = DICE_VALUE(상대주사위수, 상대주사위면, 상대수정치)
	LOCAL = 상대주사위값 - 플레이어주사위값

	;지참아이템【바렛골렘】の효과　基本ダイス差を0.8배に
	SIF LOCAL && 지참아이템 == 855
		TIMES LOCAL, 0.8

	CALL DICE_PRINT(플레이어주사위값, 0, 0, 상대주사위수, 상대주사위면, 상대수정치, "確定後")
	$BEFORE_BATTLE
	PRINTL
	PRINTFORML 남은HP: {남은HP}　남은 봄 사용횟수: {남은폭탄횟수}　봄 위력: Min:{B_Min}, Max:{B_Max}　남은잔기: {플레이어잔기}

	IF 지참아이템
		SETCOLOR 0x666666
		PRINTFORML (아이템 효과: %ITEMNAME:지참아이템% …%DANMAKU_ITEM_EXPLANATION(지참아이템)%)
	ENDIF

	IF TCVAR:MASTER:국사무쌍약
		SETCOLOR 0x666666
		PRINTFORML   （효과발동중: 국사무쌍약　　…주사위면이 미세하게 좋아졌다）
	ENDIF

	IF TCVAR:MASTER:인챈트
		SETCOLOR 0x666666
		PRINTFORML   （버프 효과: 인챈트　　…주사위 수정치 Up）
	ENDIF
	IF TCVAR:MASTER:가호
		SETCOLOR 0x666666
		PRINTFORML   （버프 효과: 가호　　…상대의 주사위 수정치 Down）
	ENDIF
	IF TCVAR:OPPONENT:인챈트
		SETCOLOR 0xc86666
		PRINTFORML   （상대의 버프 효과: 인챈트　　…상대의 주사위 수정치 Up）
	ENDIF
	IF TCVAR:OPPONENT:가호
		SETCOLOR 0xc86666
		PRINTFORML   （상대의 버프 효과: 가호　　…주사위 수정치 Down）
	ENDIF
	RESETCOLOR

	CALL ASK_M("회피에 전념한다", 1, "승부한다", 1)
	SELECTCASE RESULT
		CASE 0
			;【영력충전】の효과
			IF TALENT:MASTER:영력충전 == 1
				영력충전플래그 = 1
				CALL COLORMESSAGE(@"%CALLNAME:MASTER%의【영력충전】발동!",C_RED,2)
			ENDIF
			PRINTFORML 날아오는 탄막을 회피하는 것에 전념하기로 했다.
			PRINTFORMW ...
			PRINTL
			CALL DICE_PRINT(플레이어주사위값, 상대주사위값, 0, 0, 0, 0, "勝負")
			PRINTW
			;【바렛골렘】の효과  回避選択時はさらに0.9배（トータル0.72배程度に）
			IF LOCAL && 지참아이템 == 855
				TIMES LOCAL, 0.9
				PRINTFORML 골렘이 방패가 되어 탄막을 막았다…！
			ENDIF
			$UseEpitaph
			;【그레이즈능숙함】の효과
			IF TALENT:MASTER:그레이즈능숙함 == 1
				IF ABL:MASTER:전투능력 < 3
					LOCAL = (LOCAL * 2) / (ABL:MASTER:전투능력 + 2)
				ELSE
					LOCAL = (LOCAL * 3) / (ABL:MASTER:전투능력 + 3)
				ENDIF
			ENDIF
			DamageThreshold = MAX(ABL:MASTER:전투능력 / 2, 1)
			IF LOCAL <= 20 * DamageThreshold
				PRINTFORML 살짝 긁혔을 뿐이다.
				SIF LOCAL > 0
					남은HP -= LOCAL * 4
			ELSEIF LOCAL <= 50
				PRINTFORML 약간 타격을 입었다.
				남은HP -= LOCAL * 5
			ELSEIF LOCAL <= 150
				PRINTFORML 꽤 타격을 입었다.
				남은HP -= LOCAL * 6
			ELSE
				PRINTFORML 상당한 타격을 입었다.
				남은HP = 남은HP - MAXBASE:MASTER:체력 / 2 + 1
			ENDIF
			PRINTFORMW HP: {남은HP}
			IF 남은HP <= 0
				Pichun = 1
			ENDIF
		CASE 1
			;【乾坤一擲】の효과
			IF GETBIT(CFLAG:OPPONENT:탄막특능, 7)
				IF RAND:100 + 1 < 10 + 상대세기 * 2
					LOCAL:1 = 상대세기 * (7 + RAND:5) * (상대세기 / 2 + 1) + RAND:7
					상대주사위값 += LOCAL:1
					SETCOLOR C_RED
						PRINTFORML %CALLNAME:OPPONENT%의【건곤일척】발동!
						PRINTFORMW %CALLNAME:OPPONENT%의 주사위 값이 {LOCAL:1} 만큼 증가했다!
					RESETCOLOR
					;口上呼び出し
					CALL KOJO_MESSAGE_SEND("DANMAKU", 상대잔기, OPPONENT, ENEMY, -1, "乾坤一擲")
					LOCAL = 상대주사위값 - 플레이어주사위값
				ENDIF
			ENDIF
			IF 남은폭탄횟수 >= 1
				PRINTFORML 봄을 사용합니까?
				CALL ASK_M("사용한다", 1, "사용하지 않는다", 1)
				SELECTCASE RESULT
					CASE 0
						CALL UseBomb
						BombPower = RESULT
						플레이어주사위값 += BombPower
						PRINTFORMW 주사위 값이 {BombPower} 만큼 증가했다!
						남은폭탄횟수 --
						LOCAL = 상대주사위값 - 플레이어주사위값
					CASE 1

				ENDSELECT
			ENDIF
			PRINTFORML 승부다!
			PRINTFORMW ...
			PRINTL
			CALL DICE_PRINT(플레이어주사위값, 상대주사위값, 0, 0, 0, 0, "勝負")
			IF (지참아이템 == 870 ||지참아이템 == 871) && LOCAL > 0 && BASE:MASTER:TSP >= 500
				SETBGCOLOR 1, 24, 60
				{
				CALL COLORMESSAGE(@"%TEXTR("보인다！　이녀석의 움직임이！　뭘 하려는 건지, 완전히！
				/『읽을 수 있다』………움직임의 『궤적』을 읽을 수 있다……『미래로의 움직임의 궤적』이…！")%", 0xFF1493, 2, 1)
				}
				PRINTFORML %마스터는% 『에피타프』를 통해 자기 자신의 피탄을 예언했다！
				CALL RECOVER(MASTER, -500, "TSP")
				CALL COLORMESSAGE(@"남은TSP：{BASE:MASTER:TSP}", C_RED, 2)
				CALL ASK_M("회피에 전념한다",1,"『킹・크림슨』！(TSP1500)",지참아이템 == 871 && BASE:MASTER:TSP >= 1500)
				SELECTCASE RESULT
					CASE 0
						PRINTFORML %마스터는% 순간적으로 회피로 전환했다！
						RESETBGCOLOR
						GOTO UseEpitaph
					CASE 1
						CLEARLINE 1
						PRINTL
						SETCOLOR 0xFF1493
						{
						PRINTFORML %TEXTR("『킹・크림슨』의 능력 속에서 이 세상의 시간은 지워지고……그리고 모든 인간은 이 시간 속에서 흘러간 족적을 기억하지 못한다！
						/『하늘의 구름은 제 몸이 뜯겨 나간 것을 모르나니！』『꺼진 불꽃은 그 자신조차 꺼진 순간을 인식하지 못한다！』/『결과』뿐이다！！이 세상은 『결과』만이 남는다！！
						/시간이 날아간 세계에서 『움직임』은 모두 무의미해지는 것이다！！")%
						}
						PRINTFORMW 이것이 내 『킹・크림슨』의 능력！『시간을 지우고』뛰어넘는다……！！
						RESETCOLOR
						RESETBGCOLOR
						CALL RECOVER(MASTER, -1500, "TSP")
						PRINTFORML %CALLNAME:MASTER%의 후방으로、탄막이 날아간다――
						IF RAND:3 + 4 < 상대세기
							PRINTFORMW 계속 %CALLNAME:MASTER%의 불가능한 각도에서의 반격을、%조사처리(CALLNAME:OPPONENT,"는")% 흘린 핏방울의 수로 알아차려 회피했다！
							LOCAL = 0
						ELSE
							PRINTFORMW 경악에 눈을 크게 뜰 새도 없이、%조사처리(CALLNAME:OPPONENT,"는")% 사각에서 이해할 수 없는 반격을 받았다！
							LOCAL = -100
						ENDIF
				ENDSELECT
			ENDIF
			PRINTW
			IF LOCAL > 0
				IF 남은폭탄횟수 >= 1
					PRINTFORML 피할 길이 없어져 버린 %마스터는% 봄을 사용했다!
					CALL UseBomb
					BombPower = RESULT

					SETCOLOR C_RED
					PRINTFORM 주사위값의 차:{LOCAL}
					IF LOCAL <= BombPower
						PRINTFORML ≤ {BombPower}
						PRINTFORMW %마스터는% 봄으로 %CALLNAME:OPPONENT%의 탄막을 소거했다!
						LOCAL = 0
					ELSE
						PRINTFORMW ＞ {BombPower}
					ENDIF
					RESETCOLOR
					남은폭탄횟수 --
				ENDIF
			ELSEIF LOCAL < 0
				;【怪力乱神】の효과
				IF GETBIT(CFLAG:OPPONENT:탄막특능, 4) && (-LOCAL) < 상대세기 * 10
					SETCOLOR C_RED
						PRINTFORML %CALLNAME:OPPONENT%의 【괴력난신】발동!
						PRINTFORMW %CALLNAME:MASTER%의 탄막이 %조사처리(CALLNAME:OPPONENT,"를")% 피탄시키기 전에 소거되었다!
						;口上呼び出し
						CALL KOJO_MESSAGE_SEND("DANMAKU", 상대잔기, OPPONENT, ENEMY, -1, "怪力乱神")
						LOCAL = 0
					RESETCOLOR
				ENDIF
			ENDIF
			IF LOCAL < 0
				CALL COLORMESSAGE(@"격파했다! 해냈어!",C_YELLOW,1)
				상대잔기 --
				;口上呼び出し
				CALL KOJO_MESSAGE_SEND("DANMAKU", 상대잔기, OPPONENT, ENEMY, -1, "被弾")
				PRINTFORMW 상대 잔기: {상대잔기}
			ELSEIF LOCAL == 0
				PRINTFORMW 서로 탄막을 맞출수 없었다.
			ELSE
				Pichun = 1
			ENDIF
		CASE 2

	ENDSELECT
	IF Pichun
		PRINTFORML ＼피츙／
		PRINTFORML 피탄당하고 말았다......
		플레이어잔기 --
		남은HP = MAXBASE:MASTER:체력
		남은폭탄횟수 = 2
		PRINTFORMW LIFES: {플레이어잔기}
		;【루나슈터】の효과
		SIF TALENT:MASTER:루나슈터 == 1
			남은폭탄횟수 ++
		;【残忍酷薄】の효과
		IF GETBIT(CFLAG:OPPONENT:탄막특능, 6) && 플레이어잔기 > 0
			CALL COLORMESSAGE(@"%CALLNAME:OPPONENT%의 【잔인무도】발동!",C_RED,2)
			;【세방울의천수】の효과
			IF ITEM:세방울의천수 && !RAND:2
				PRINTFORMW 개구리 신의 가호로【잔인무도】효과를 방어했다!
			ELSE
				;口上呼び出し
				CALL KOJO_MESSAGE_SEND("DANMAKU", 상대잔기, OPPONENT, ENEMY, -1, "残忍酷薄")
				남은폭탄횟수 --
			ENDIF
		ENDIF
	ENDIF
	Pichun = 0
	턴수 ++
WEND

PRINTL
IF 플레이어잔기 <= 0
	PRINTFORMW %마스터는% 져 버렸다……
	CALL COLORMESSAGE(@"패배자!",C_RED,2,1)
	TCVAR:OPPONENT:탄막승부결과 = -1
ELSEIF 상대잔기 <= 0
	PRINTFORML %마스터는% %조사처리(CALLNAME:OPPONENT,"와")%의 승부에서 이겼다!
	CALL COLORMESSAGE(@"인생의 승리자!",C_GREEN,2,1)
	TCVAR:OPPONENT:탄막승부결과 = 1
	CALL DANMAKU_GIPS_REWARD
ENDIF
TFLAG:GipsLV = 0
PRINTL

;楽譜入手処理
IF (RAND:(100) < 30)
	RESULT = 악보입수_탄막승부(OPPONENT)
	SIF RESULT != -1
		PRINTL
ENDIF

RETURN 상대잔기

@DICE_PRINT(ARG:0, ARG:1, ARG:2, ARG:3, ARG:4, ARG:5, ARGS)
IF ARGS == "確定前"
	PRINTFORM %CALLNAME:MASTER%:
	SETCOLOR C_GREEN
	PRINTFORM {ARG:0}d{ARG:1}
	IF ARG:2 > 0
		SETCOLOR C_RED
		PRINTFORM +{ARG:2}
	ELSEIF ARG:2 < 0
		SETCOLOR C_BLUE
		PRINTFORM {ARG:2}
	ENDIF
	RESETCOLOR
	PRINTFORM (Min:{ARG:0 + ARG:2}, Max:{ARG:0 * ARG:1 + ARG:2})  VS
	SETCOLOR C_GREEN
	PRINTFORM {ARG:3}d{ARG:4}
	IF ARG:5 > 0
		SETCOLOR C_RED
		PRINTFORM +{ARG:5}
	ELSEIF ARG:5 < 0
		SETCOLOR C_BLUE
		PRINTFORM {ARG:5}
	ENDIF
	RESETCOLOR
	PRINTFORM (Min:{ARG:3 + ARG:5}, Max:{ARG:3 * ARG:4 + ARG:5})
	PRINTFORML :%CALLNAME:TARGET%
;ARG:1,2は使わない
ELSEIF ARGS == "確定後"
	PRINTFORM %CALLNAME:MASTER%:
	SETCOLOR C_GREEN
	PRINTFORM {ARG:0}
	RESETCOLOR
	PRINTFORM VS
	SETCOLOR C_GREEN
	PRINTFORM {ARG:3}d{ARG:4}
	IF ARG:5 > 0
		SETCOLOR C_RED
		PRINTFORM +{ARG:5}
	ELSEIF ARG:5 < 0
		SETCOLOR C_BLUE
		PRINTFORM {ARG:5}
	ENDIF
	RESETCOLOR
	PRINTFORM (Min:{ARG:3 + ARG:5}, Max:{ARG:3 * ARG:4 + ARG:5})
	PRINTFORML :%CALLNAME:TARGET%
;ARG:2,3,4,5は使わない
ELSEIF ARGS == "勝負"
	PRINTFORM %CALLNAME:MASTER%:
	SETCOLOR C_GREEN
	PRINTFORM {ARG:0}
	RESETCOLOR
	PRINTFORM VS
	SETCOLOR C_GREEN
	PRINTFORM {ARG:1}
	RESETCOLOR
	PRINTFORML :%CALLNAME:TARGET%
ENDIF

@DICE_VALUE(ARG:0, ARG:1, ARG:2)
#FUNCTION
;初期化しないとダメ
LOCAL:1 = 0
FOR LOCAL, 0, ARG:0
	LOCAL:1 += RAND:(ARG:1) + 1
NEXT

RETURNF LOCAL:1 + ARG:2

@SHOW_DANMAKU_RULE
PRINTFORML 서로 주사위를 던져 나온 눈의 크기를 겨룬다는 게 기본 룰입니다
PRINTFORMW 처음은 서로 잔기를 3개 보유하고 있으며, 전투능력이 클수록 주사위 수와 면이 증가합니다.
PRINTFORMW 플레이어만, 자신의 주사위눈을 본 다음 회피에 전념할지 승부할지를 선택할 수 있습니다(이때, 상대의 주사위눈은 보이지 않습니다).
PRINTFORMW 회피에 전념하기로 선택했을 경우는, 잔기가 줄어들지 않는 대신에 주사위눈 차이에 따라 HP에 대미지를 받습니다.
PRINTFORMW HP가 0이 되면 플레이어의 잔기가 1 줄어듭니다. 잔기가 0이 되면 플레이어의 패배입니다.
PRINTFORMW 승부하기로 선택하고 상대보다 낮은 주사위눈이었을 경우는, 무조건 플레이어의 잔기가 1 줄어듭니다.
PRINTFORMW 승부하기로 선택하고 상대보다 높은 주사위눈을 냈을 때에만 상대의 잔기를 1 줄일 수 있습니다.
;残機0ならまだプレイできるじゃんとか言わない
PRINTFORMW 플레이어의 잔기가 0이 되기 전에 상대의 잔기를 0으로 만들면 플레이어의 승리가 됩니다.
PRINTFORMW 또한 플레이어는 폭탄을 사용해 주사위의 결과에 플러스할 수 있으므로, 이때다 싶은 순간에 사용합시다.
PRINTFORMW 더해서 주사위 승부에 패배해 플레이어의 잔기가 줄어들 때, 폭탄을 사용해 잔기가 줄어드는 것을 막을 때가 있습니다.
PRINTFORMW 단, 잔기가 줄어들면 폭탄 사용 횟수가 강제적으로 2가 되므로, 사용하지 않은 채로 잔기가 줄면 손해를 봅니다.

@UseBomb
#DIM BombPower
BombPower = ABL:MASTER:전투능력 * (7 + RAND:5) * (ABL:MASTER:전투능력 / 2 + 1) + RAND:7

CALL COLORMESSAGE(@"봄 위력 {BombPower}",C_YELLOW,1)
;【루나슈터】の효과
IF TALENT:MASTER:루나슈터 == 1
	BombPower = BombPower * (RAND:3 + 2)
	CALL COLORMESSAGE(@"【루나슈터】의 효과로 봄 위력이 {BombPower} 만큼 증가했다!",C_YELLOW,2)
ENDIF
;【세방울의천수】の효과
IF ITEM:세방울의천수
	;LOCAL = MAX(RAND:750, 400)
	LOCAL = BombPower / 10
	BombPower += LOCAL
	CALL COLORMESSAGE(@"%CALLNAME:MASTER%의 주머니에서 조용히 수많은 개구리들이 튀어나왔다!",C_YELLOW,1)
	CALL COLORMESSAGE(@"봄 위력이 {LOCAL} 만큼 증가했다!",C_YELLOW,1)
ENDIF
RETURN BombPower

@DANMAKU_ITEM_USE
#DIM 지참아이템
#DIMS 효과텍스트
지참아이템 = 0
DRAWLINE
PRINTFORML 탄막승부 보조 아이템을 사용할까요?
DRAWLINE
FOR LOCAL,850, 900
	SIF !DANMAKU_ITEM_AVAILABLE(LOCAL)
		CONTINUE
	PRINTFORML [{LOCAL}] - %ITEMNAME:LOCAL%
NEXT
DRAWLINE
PRINTFORML [999] - 사용하지 않는다
INPUT
IF !DANMAKU_ITEM_AVAILABLE(RESULT)
	RETURN 0
ENDIF

지참아이템 = RESULT
PRINTFORML %조사처리(ITEMNAME:지참아이템,"을")% 사용합니까？
PRINTFORML %DANMAKU_ITEM_EXPLANATION(지참아이템)%
CALL ASK_YN
IF !RESULT
	;先に進む
ELSE
	RESTART
ENDIF
ITEM:지참아이템 -= 1
RETURN 지참아이템

@DANMAKU_ITEM_EXPLANATION(ARG)
#FUNCTIONS
#DIMS 효과텍스트
SELECTCASE ARG
	CASE 850
		효과텍스트 = 주사위면이 어렴풋이 좋아진다
	CASE 851
		효과텍스트 = 주사위수가 어렴풋이 증가한다
	CASE 852
		효과텍스트 = 주사위면이 조금 좋아진다
	CASE 853
		효과텍스트 = 주사위수가 조금 증가한다
	CASE 854
		효과텍스트 = 주사위면이 조금 좋아지고、주사위수가 조금 증가한다
	CASE 855
		효과텍스트 = 받은 데미지를 조금 경감한다
	CASE 856
		효과텍스트 = 잔기가 １ 늘어난다
	CASE 857
		효과텍스트 = 봄 수가 １ 늘어난다
	CASE 870
		효과텍스트 = 주사위에서 지더라도 회피를 택한 것이 된다　발동마다 TSP500소비
	CASE 871
		효과텍스트 = 피탄을 무효화하며 상대의 잔기를 줄이고 치트 능력이지만 TSP소비가엄청나게 크다
	CASEELSE
		효과텍스트 =
ENDSELECT
RETURNF 효과텍스트

@DANMAKU_ITEM_EFFECT(ARG)
#DIM Dnum
#DIM Dface
Dnum = 0
Dface = 0
SELECTCASE ARG
	CASE 850	;폭발성시약
		Dface = MAX(RAND:((ABL:MASTER:전투능력) + 3), 2)
	CASE 851	;아이시클봄
		Dnum = MAX(RAND:((ABL:MASTER:전투능력) + 1), 1)
	CASE 852	;태을의불씨
		Dface = MAX(RAND:((ABL:MASTER:전투능력) * 2), 4)
	CASE 853	;매혹의미끼
		Dnum = MAX(RAND:((ABL:MASTER:전투능력) * 2), 3)
	CASE 854	;현자의돌
		Dface = MAX(RAND:((ABL:MASTER:전투능력 + 1) * 3), 6)
		Dnum = MAX(RAND:((ABL:MASTER:전투능력 + 1) * 2), 5)
	CASE 855	;바렛골렘
	CASE 856	;命のカード
	CASE 857	;スペルのカード
	CASE 870	;에피타프
	CASE 871	;キンクリ
ENDSELECT
RETURN Dface, Dnum

@DANMAKU_ITEM_AVAILABLE(ARG)
#FUNCTION
SIF !INRANGE(ARG, 850, 900)
	RETURNF 0
SIF DANMAKU_ITEM_EXPLANATION(ARG) == ""
	RETURNF 0
SIF !ITEM:ARG
	RETURNF 0
RETURNF 1

@DANMAKU_GipsEquip(ARGS)
IF ARGS == "椛口上" || ITEM:루나슈터양성깁스
	PRINTFORML %마스터는% 루나슈터양성깁스를 장비하고、버클의 손잡이에 손을 대었다
	CALL ASK_M("역시 그만둔다", 1, "Easy",1, "Normal", 1, "Hard", 1, "Lunatic" , 1, "『위험・수동모드』", 1 )
ELSE
	PRINTFORML %마스터는% 슈터깁스를 장비하고、버클의 손잡이에 손을 대었다
	CALL ASK_M("역시 그만둔다", 1, "Easy",1, "Normal", 1, "Hard", 1)
ENDIF
SELECTCASE RESULT
	CASE 0
		RETURN
	CASE 1
		TFLAG:GipsLV = 1
	CASE 2
		TFLAG:GipsLV = 3
	CASE 3
		TFLAG:GipsLV = 5
	CASE 4
		TFLAG:GipsLV = 10
	CASE 5
		PRINTFORML 부하 레벨을 설정해주세요
		PRINTL 참고 부하 수준　Easy:1　Normal:3　Hard:5　Lunatic:10
		PRINTL [0] [1] [3] [5] [10] [15] [30] [50] [100]
		INPUT
		TFLAG:GipsLV = MAX(RESULT, 0)
ENDSELECT

@DANMAKU_GIPS_REWARD
#DIM Reward
SIF TFLAG:GipsLV <= 0
	RETURN
PRINTFORML 깁스의 효과로 %마스터는% 넉넉한 경험치를 받았다
Reward = MIN(15, TFLAG:GipsLV)
;SELECTCASE GipsLV
;	CASE IS >= 10
;		Reward = MIN(15, GipsLV)
;	CASE IS >= 5
;		Reward = 4
;	CASE IS >= 3
;		Reward = 2
;	CASEELSE
;		Reward = 1
;ENDSELECT
IF ABL:MASTER:전투능력 <= ABL:TARGET:전투능력
	IF TFLAG:GipsLV >= 10
		Reward *= 3
		CALL COLORMESSAGE(@"상대가 동격 이상이기에 추가 경험치가 3배",C_YELLOW,1)
	ELSE
		Reward *= 2
		CALL COLORMESSAGE(@"상대가 동격 이상이기에 추가 경험치가 2배",C_YELLOW,1)
	ENDIF
ENDIF
CALL AddEXP("전투경험" , MASTER, Reward)